
บทที่   Ä
กรณีศึกษาระบบงานทะเบียน

จดุประสงคเชิงพฤติกรรม
1. เขาใจระบบงานของส ํานักทะเบียน ในมุมมองของ Programmer มากขึ้น
2. เขาใจการสรางระบบงานคอมพิวเตอร โดยพิจารณาจากความตองการของผูใช
3. มปีระสบการณ และมองเห็นแนวการออกแบบเมนูทั้งระบบงาน
4. เขาใจแนวการเขียน และการแปลโปรแกรม โดยใชวิธีการอยางงาย
5. เขาใจการท ํารายงาน และการปรับปรุงขอมูลในรูปแบบตาง ๆ
6. เขาใจการเขียนเมน ูและเรียกใชงานอยางงาย

หวัขอในบทเรียน
5.1  ศึกษาความตองการ
5.2  โครงสรางขอมูล
5.3  วิเคราะหระบบงานเบื้องตน
5.4  โปรแกรมในระบบงานทะเบียน

5.4.1  โปรแกรมตาง ๆ ใน MAIN.PRG     : สารบัญของโปรแกรม
5.4.2  โปรแกรมตาง ๆ ใน SUB01.PRG  : รวมโปรแกรมเตรียมขอมูล และการลงทะเบียน
5.4.3  โปรแกรมตาง ๆ ใน SUB02.PRG  : รวมโปรแกรมรายงานหลังลงทะเบียน
5.4.4  โปรแกรมตาง ๆ ใน SUB03.PRG  : รวมรายงานตามคํ ารองของผูใช
5.4.5  โปรแกรมตาง ๆ ใน SUB04.PRG  : รวมรายงานประจํ าภาคเรียน
5.4.6  โปรแกรมตาง ๆ ใน SUB05.PRG  : รวมโปรแกรมปรับปรุงขอมูล

ตวัอยางโปรแกรมเกี่ยวกับระบบงานทะเบียน   38  ตัวอยาง

หมายเหตุ
ตัวอยางในบทนี้ประยุกตจากประสบการณที่ผูเขียนทํ างานเขียนโปรแกรมใหสํ านักทะเบียนโยนก ซึ่งไดมีการ

ประสานงานกบัคณุลัดดาวรรณ  ปญญา ซึ่งเปนผูปฏิบัติงาน แลวเรียบเรียงใหม ใหเหมาะสมกับเนื้อหาวิชา และเวลาที่
จะใชสอน ใหดเูหมือนเปนระบบที่อาจน ําไปใชได แตในความเปนจริงแลว เปนเพียงกรณีศึกษา เพื่อชี้ใหเห็นถึงจุดที่ตอง
ปรับปรุงเพิ่มเติมอีกมาก .. ประกอบการบรรยายในชั้นเรียน และโปรแกรมทั้งหมดในหนังสือ นักศึกษาสามารถ 
Download ไปศกึษา ประกอบการเรียนภาคปฏิบัติ

บทที่ 5   กรณีศึกษาระบบงานทะเบียน
เนือ่งจากระบบทะเบียนทางการศึกษา ซึ่งเก็บประวัติการเรียนของนักเรียน เปนระบบที่น ํามาใชเก็บประวัติการ

ศกึษาในทกุสถาบนั โดยแตละสถาบันมีรายละเอียดบางประการที่ไมเหมือนกัน  แตแกนแทของงานทะเบียนแลวใกล
เคยีงกนั คอืงานบริการขอมูลการศึกษาส ําหรับนักศึกษา อาจารยและผูบริหารของสถานศึกษานั้น ๆ

ดวยเหตทุีผู่สนใจศึกษาการเขียนโปรแกรมทุกคน  ตองผานการศึกษามาอยางนอยหนึ่งสถาบัน และไดเขาไป
เกีย่วของกบังานทะเบียน หรือมีรายชื่อของตนในสถาบันนั้น และไดรับรายงานเชน ใบแจงผลการเรียน  ท ําใหทุกคนได
คุนเคย และมองภาพรวมของงานทะเบียน วาระบบนี้ควรมีงานอะไรที่ตองทํ าบาง

ดงันัน้ผูเขยีนจงึของน ําระบบทะเบียนมาเปนกรณีศึกษา  เพื่อใหผูอานไดเขาใจในเนื้อหาของระบบงาน  เพื่อ
น ําไปสูหลกัการพัฒนาโปรแกรมในระบบอื่น ๆ ตอไป ซึ่งผูเขียนขอเริ่มตั้งแตการศึกษาความตองการของผูใช แลวน ํามา
วเิคราะหหาความตองการ  ที่จะน ํามาเปนแนวทางในการพัฒนาเปนระบบยอยตาง ๆ จนถึงการเขียนโปรแกรมใหได
ตามความตองการ  12

&  5.1 ศึกษาความตองการ
แผนกทะเบยีนทีจ่ะศึกษาตอไปนี ้  ไมไดเนนที่จะศึกษาจนครอบคลุมทุกอยางที่แผนกทะเบียนทํ า แตจะศึกษา

เฉพาะสวนงานที่เกี่ยวกับการลงทะเบียน จัดเก็บเกรด และรายงานเกรดใหนักศึกษาทราบ

5.1.1 บุคคลที่เกี่ยวของกับงาน
ลํ าดับ ผูเกี่ยวของ ความสัมพันธตอขอมูล
1 นกัศึกษา  มาลงทะเบียนเรียน และรอรับผลการเรียนเมื่อสิ้นภาคการศึกษา
2 อาจารยผูสอน เปนผูระบุวิชาที่ตองการสอน รอรับรายชื่อ และสงเกรดทายภาค
3 ผูบริหาร เปนผูใหนโยบาย และรับรายงานสรุปผลตาง ๆ
4 แผนกทะเบียน ท ําหนาที่ติดตอกับฝายตาง ๆ เพื่อน ําขอมูลเขาระบบ และพิมพรายงาน

5.1.2 ระบบงานที่เปนอยู และสิ่งที่ผูใชตองการ
ความตองการของผูใชมีมากมาย โดยเฉพาะการพูดคุยกับผูใชที่ไมรูจักคอมพิวเตอร จํ าเปนตองตีความหมาย

ระบบงานทีเ่ปนอยู และเรียบเรียงใหมเปนขอ ๆ  แลวใหผูใชตรวจสอบอีกครั้งหนึ่งวาสิ่งที่ผูใชตองการ กับทีมพัฒนา
ระบบเขาใจตรงกันหรือไม

บอยครัง้ทีก่ารพูดคุยระหวางทีมพัฒนาระบบ กับกลุมผูใช   ถึงปญหาในระบบงานเดิม  ทํ าใหตองปรับปรุงการ
ท ํางานแบบเดมิในหลาย ๆ จุด เพราะในการน ําระบบคอมพิวเตอรเขามาใชนั้น  การท ํางานตองมีขั้นตอนที่ชัดเจน และ
ตรวจสอบความถูกตองได   การทะเลาะและไมเขาใจกันจึงเกิดขึ้นจนเปนเรื่องปกติในระหวางการพัฒนาระบบงานใหม
แตทัง้สองฝายควรทํ าความเขาใจซึ่งกันและกัน  เพื่อความสมบูรณของระบบใหมที่จะนํ ามาใช หลังจากการพัฒนาเสร็จ
สิ้นลง

12 ระบบทะเบยีนนีอ้างอิงจากประสบการณของผูเขียน ที่ท ํางานในสํ านักทะเบียนวิทยาลัยโยนก แลวประยุกต
มาเปนรูปแบบของ Clipper ซึง่เดมิพัฒนาดวย COBOL แลวเปลี่ยนมาเปน Microsoft Access

บางครัง้จ ําเปนตองเปลี่ยนแปลงระบบงานที่เปนอยูใหมีความเปนระเบียบ   เพื่อจะไดจัดเก็บขอมูลและ
ประมวลผลในคอมพวิเตอรไดอยางถูกตอง  รายการตอไปนี้เปนความตองการของผูใชที่ไดปรับปรุงใหเขาใจไดงาย เปน
ขอ ๆ โดยศึกษามาจากระบบงานที่มีอยู
ความตองการของผูใช
1. อาจารยผูสอนใหขอมูลการเปดวิชา กอนเปดภาคการศึกษาใหม 1 เดือน
2. น ําขอมูลนักศึกษาใหมเขาฐานขอมูล กอนเปดเรียน
3. การลงทะเบียน มีการตรวจสอบจํ านวนไมใหเกินที่กํ าหนด
4. การขอแกไข บอกเพิ่ม-ลดวิชา 1 สัปดาหหลังวันลงทะเบียน
5. พมิพขอมูลการลงทะเบียนติดประกาศ ใหนักศึกษาตรวจสอบอีกครั้งหนึ่ง
6. พมิพรายชื่อนักเรียนใหอาจารยผูสอน
7. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
8. การขอเพิกถอนวิชา ของนักเรียน
9. พมิพรายชื่อแตละวิชาติดหนาหองสอบ และเลขที่โตะของนักศึกษา
10. การสงเกรดของอาจารย เมื่อสิ้นภาคเรียน
11. พมิพใบรายงานผลการเรียนประจํ าภาค
12. ปรบัปรงุขอมูลอาจารย เมื่อรับอาจารยใหม ลาออก หรือแตงงาน
13. พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
14. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช
15. แสดงตารางเรียนของนักศึกษา เมื่อผูปกครองมาขอพบ
16. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
17. พมิพรายชื่อแยกตามสถานภาพ
18. พมิพรายชื่อแยกตามสาขา
19. พมิพใบรายงานผลการเรียนตลอดหลักสูตร เมื่อนักศึกษารองขอ
20. การปรบัปรุงและพิมพขอมูลจากแฟมมาตรวจสอบ

&  5.2 โครงสรางขอมูล
เมือ่ทราบถงึสิ่งที่ตองการ  ทีมพัฒนาระบบจะออกแบบแฟมขอมูลส ําหรับเก็บขอมูลหลาย ๆ แฟมซึ่งรวมกัน

แลวเรยีกวาฐานขอมูล ซึ่งจะเก็บขอมูลเพื่อรองรับกับสิ่งที่ผูใชตองการ  จึงตองมีการลองผิดลองถูก หากกลุมแฟมที่จัด
ขึน้ยงัไมสามารถเก็บขอมูลเพื่อสนองสิ่งที่ผูใชตองการได จะตองแกไขใหมจนกวาจะไดกลุมแฟมที่รองรับงานของผูใชได
หมด

แฟมที่เกี่ยวของกับระบบงานทะเบียน
1. แฟมนักศึกษา   IDSTD.DBF
   1.1 รหัส             *IDSTD         9(7) ตวัอยางขอมูล
      (2หลกัแรก:ปที่เขา 2หลักตอมา:คณะที่เรียน 3หลักสุดทาย:ล ําดับที่) 3901001 นาย สมคิด  สมควร  11 A
  1.2 ชื่อ              INAME         X(20) 3901002 น.ส.ถนัด   ประดิษฐ 14  A
  1.3 สกุล             ISURN         X(20) 3901003 นาย สมบัต ิภูเบศร   12  D
  1.4 สาขา              IMAJOR        99
      (11:บญัช ี12:การจัดการ 13:การตลาด 14:คอมพิวเตอรธุรกิจ)
  1.5 สถานภาพ           ISTAT         X
      (A:ก ําลังศึกษาอยู   D:ลาออก)
2.  แฟมวิชา     SUBJECT.DBF
  2.1 ชือ่วิชา         *SUBJ          X(8) ตวัอยางขอมูล
     (4หลกัแรก:ชื่อหมวด 3หลักหลัง:ชื่อล ําดับของวิชาในหมวด) SOCI 101 3930005 4201 09.00-11.50 28 30
  2.2 รหัสผูสอน        TEACHID       9(7) BCOM101 3914007 1203 08.00-10.50 6 6
(2หลกัแรก:ปที่เขา  ตอมา:หนวยงานที่ประจํ า 3หลักสุดทาย:ล ําดับ) ENGL202  3930002 1211 14.00-16.50 25 25
  2.3 เลขที่หอง        ROOM          X(4)
      (เบอรหองที่ปรากฏหนาหองตามตึกตาง ๆ)
  2.4 เวลาเรียน         STIME         X(11)
      (เชน 09.00-11.50 เริ่มเรียน จนถึงเลิกเรียน)
  2.5 จ ํานวนลงทะเบียน    SREGIST       999
  2.6 จ ํานวนรับได       SLIMIT        999
3.  แฟมอาจารยผูสอน   TEACHER.DBF
  3.1 รหัสผูสอน       *TEACHID       9(7) ตวัอยางขอมูล
 (2หลกัแรก:ปที่เขา  ตอมา:หนวยงาน  3หลักสุดทาย:ล ําดับที่) 3910001 นาย อติชาติ     ปรีชา         40  M
  3.2 ชื่อผูสอน        TNAME         X(20) 3914007 น.ส.ศริิลักษณ  เกศรินทร     45  F
  3.3 สกุลผูสอน        TSURN         X(20) 3930005 นาย จักรพันธ   สุวรรณ       34  M
  3.4 อายุ             TAGE          999
  3.5 เพศ               TSEX          X

(M:ผูชาย F:ผูหญิง)

4.  แฟมลงทะเบียน   REGIST.DBF
4.1 ปการศึกษา       *RYEAR         9999  ตวัอยางขอมูล
4.2 ภาคเรียน         *RSEM          X 2002  1  3901001   BCOM 101   4.0
4.3 รหัสนักศึกษา     *IDSTD         9(7) 2002  1  3901001   SOCI 101     2.5
4.4 ชือ่วิชา         *SUBJ          X(8) 2002  1  3901001   ENGL 101    2.0
4.5 เกรดที่ไดรับ     RGRADE        9V9 2002  1  3901002   BCOM 101   1.5
(0:F  1:D  1.5:D+  2:C  2.5:C+  3:B  3.5:B+  4:A) 2002  1  3901002   ENGL 201    1.0

 2002  1  3901003   BCOM 301   3.5
2002  1  3901003   MATH 303   3.0
2002  2  3901001   GOVT 101    3.0
2002  2  3901001   ENGL 102    3.5
2002  2  3901003   ARTS 201    1.0
2002  2  3901003   COMM 401  1.5
2002  2  3901003   BCOM 305   2.0

 5.  แฟมหอง    ROOM.DBF
5.1 เลขที่หอง       *ROOM          X(4) ตวัอยางขอมูล
5.2 ล ําดับโตะ         RCSEQ         999  1201  1   A1
5.3 เลขที่โตะ        RCHAIR        XX  1201  2   A2
(แตละหองเลขโตะเรียงไมเหมือนกัน เชน A1,A2,A3,...)  1201  3   A3

  1201  4   B1
1201  5   B2
1202  1   A1
1202  2   A2
1202  3   B1

&  5.3  วิเคราะหระบบงานเบื้องตน
การวเิคราะหระบบงานเบื้องตน ในครั้งนี้จะสนใจในเรื่องของสิ่งที่ผูใชตองการ แลวน ํามาแยกเปนระบบเล็ก

เพือ่ทีจ่ะจบักลุมใหกระชับ และน ําไปออกแบบเมน ู แลวน ําไปเปนแนวทางในการจัดทํ าโปรแกรม  แตหากนํ าไปใชงาน
จรงิจะตองมกีารศึกษาความเปนไปได  ผังงาน ความสัมพันธระหวางขอมูล และตนทุนใหชัดเจนกวานี้

5.3.1 จัดระบบยอยจากความตองการของผูใช
ระบบเตรียมงานและงานลงทะเบียน
   1. อาจารยผูสอนใหขอมูลการเปดวิชา กอนเปดภาคการศึกษาใหม 1 เดือน
   2. ปรบัปรงุขอมูลนักศึกษาเดิม และน ําขอมูลนักศึกษาใหมเขาฐานขอมูล กอนเปดเรียน
   3. การลงทะเบียน มีการตรวจสอบจํ านวนไมใหเกินที่กํ าหนด
   4. การขอแกไข บอกเพิ่ม-ลดวิชา 1 สัปดาหหลังวันลงทะเบียน
ระบบงานหลังลงทะเบียน
   1. พมิพขอมูลการลงทะเบียนติดประกาศ ใหนักศึกษาตรวจสอบอีกครั้งหนึ่ง
   2. พมิพรายชื่อนักเรียนใหอาจารยผูสอน
   3. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
   4. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช
   5. พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
   6. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
ระบบงานเมื่อถูกรองขอ
   1. การขอเพิกถอนวิชา ของนักเรียน
   2. แสดงตารางเรียนของนักศึกษา เมื่อผูปกครองมาขอพบ
   3. พมิพใบรายงานผลการเรียนตลอดหลักสูตร เมื่อนักศึกษารองขอ
   4. ปรบัปรงุขอมูลอาจารย เมื่อรับอาจารยใหม ลาออก หรือแตงงาน
ระบบงานประจํ าภาคเรียน
   1. พมิพรายชื่อแตละวิชาติดหนาหองสอบ และเลขที่โตะของนักศึกษา
   2. พมิพรายชื่อแยกตามสถานภาพ
   3. พมิพรายชื่อแยกตามสาขา
   4. การสงเกรดของอาจารย เมื่อสิ้นภาคเรียน
   5. พมิพใบรายงานผลการเรียนประจ ําภาค
ระบบปรับปรุงและพิมพขอมูลจากแฟม
   1. แฟมนักศึกษา 2. แฟมวิชา
   3. แฟมผูสอน 4. แฟมลงทะเบียน
   5. แฟมหอง

5.3.2 ออกแบบเมนูตามระบบยอย
เมนูหลัก
1. ระบบเตรียมงานและงานลงทะเบียน
2. ระบบงานหลังลงทะเบียน
3. ระบบงานเมื่อถูกรองขอ
4. ระบบงานประจํ าภาคเรียน
5. ระบบปรับปรุงและพิมพขอมูลจากแฟม
6. ขอมูลระบบ และทีมพัฒนาโปรแกรม
7. ออกไประบบปฏิบัติการชั่วคราว
8. เลกิการท ํางาน
1. เมนรูะบบเตรียมงานและงานลงทะเบียน
   1. รบัขอมูลการเปดวิชาใหม
   2. รบัขอมูลขอมูลนักศึกษาใหม
   3. รบัลงทะเบียนจากนักศึกษา
   4. รบับอกเพิ่ม-ลดวิชา
   5. กลับเมนูหลัก
2. เมนูระบบงานหลังลงทะเบียน
   1. พมิพขอมูลการลงทะเบียนติดประกาศ
   2. พมิพรายชื่อนักเรียนใหอาจารยผูสอน
   3. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
   4. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช
   5. พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทาน
   6. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
   7. กลับเมนูหลัก
3. เมนรูะบบงานเมื่อถูกรองขอ
   1. การขอเพิกถอนวิชา
   2. ตรวจสอบตารางเรียนของนักศึกษาเฉพาะคน
   3. พมิพรายงานผลการเรียนตลอดหลักสูตร
   4. ปรับปรุงขอมูลอาจารย
   5. กลับเมนูหลัก

4. เมนรูะบบงานประจ ําภาคเรียน
   1. พมิพรายชื่อติดหนาหองสอบ
   2. พมิพรายชื่อแยกตามสถานภาพ
   3. พมิพรายชื่อแยกตามสาขา
   4. ใสเกรด
   5. พมิพรายงานผลการเรียนประจ ําภาค
   6. กลับเมนูหลัก
5. เมนูระบบปรับปรุงและพิมพขอมูลจากแฟม
   1. แฟมนักศึกษา
   2. แฟมวิชา
   3. แฟมผูสอน
   4. แฟมลงทะเบียน
   5. แฟมหอง
   6. กลับเมนูหลัก

5.3.3 หลักการตั้งชื่อโปรแกรม
การตัง้ชือ่โปรแกรมมีหลายวิธ ีและแตละวิธีมีจุดเดน และจุดดอยที่แตกตางกัน ขึ้นกับลักษณะหนวยงาน

ลกัษณะบุคลากร และขอตกลงของทีมพัฒนาโปรแกรม
หลกัในการตั้งชื่อมีหลายวิธี
1. ตัง้ชือ่ตามลักษณะงานที่ใช
2. ตัง้ชือ่ตามหนวยงานที่เกี่ยวของ และขอมูลที่เกี่ยวของ
3. ตัง้ชือ่ตามลํ าดับในเมนู
4. ตัง้ชื่อตามผูรับผิดชอบโปรแกรม
5. ตัง้ชือ่ตามวันที่ที่เริ่มเขียนโปรแกรม
6. ตัง้ชือ่ตามลํ าดับของโปรแกรม
7. อื่น ๆ อีกมาก

   ส ําหรบัระบบงานทะเบียน ขอตั้งชื่อตามลํ าดับของตัวเลือกในเมนู เพื่อสื่อใหเขาใจไดโดยงาย   ส ําหรับชื่อของ
โปรแกรมกํ าหนดไว 5 หลัก โดยแตละหลักมีความหมายดังนี้

หลกัที ่1 ใหเปนตัว R ซึ่งยอมาจาก REGISTRA
หลักที ่2 และ 3 เปนล ําดับในเมนูหลัก
หลกัที ่4 และ 5 เปนล ําดับในเมนูยอย

โปรแกรมตาง ๆ ในระบบทะเบียน
MAIN.PRG : เมนหูลกัส ําหรับระบบงานทะเบียน เพื่อท ําตัวเลือกหลักทั้งหมด ประกอบดวย 21 โปรแกรมยอย
MAINALL : แสดงเมนูหลักบนจอภาพ
SUB1 : แสดงเมนูยอย ระบบเตรียมงาน และงานลงทะเบียน
SUB2 : แสดงเมนูยอย ระบบงานหลังลงทะเบียน
SUB3 : แสดงเมนูยอย ระบบงานเมื่อถูกรองขอ
SUB4 : แสดงเมนูยอย ระบบงานประจ ําภาคเรียน
SUB5 : แสดงเมนูยอย ระบบปรับปรุง และพิมพขอมูลจากแฟม
WORKSYSTEM : แสดงขอมูลของระบบทางจอภาพ
ASKPRT : สอบถามการปลายทางของผลลัพธ
ASKUPD : ถามยํ ้าการปรับปรุงขอมูล
SORTREG1 : จดัเรยีงแฟม REGIST ตามฟลด RYEAR,RSEM,IDSTD,SUBJ
SORTREG2 : จดัเรียงแฟม REGIST ตามฟลด SUBJ,RYEAR,RSEM,IDSTD
SORTSUB1 : จดัเรียงแฟม SUBJECT ตามฟลด ROOM,SUBJ
SORTSUB2 : จดัเรียงแฟม SUBJECT ตามฟลด SUBJ
SORTID1 : จดัเรียงแฟม IDSTD ตาม IDSTD
SORTID2 : จดัเรยีงแฟม IDSTD ตาม ISTAT,IDSTD
SORTID3 : จดัเรียงแฟม IDSTD ตาม IMAJOR,INAME,ISURN
GRADECHR : แปลงเกรดจากตัวเลขเปนตัวอักษร
GRADENUM : แปลงเกรดจากตัวอักษรเปนตัวเลข
HELPFORM : แสดงขอความชวยเหลืออยางงาย ๆ เมื่อกดปุมฟงกชัน F1
HELPREADVAR : แสดงคาที่เปนไปไดของฟลดที่ก ําลังรอรับคา
QUITPROC : สัง่เลกิท ํางานของโปรแกรมทั้งหมด
SUB01.PRG : ระบบเตรียมงานและลงทะเบียน  ประกอบดวย 4 โปรแกรมยอย
R0101 : ปรบัปรงุแฟมวิชา ดวยขอมูลการเปดวิชา จากอาจารยผูสอน
R0102 : เพิม่รายชือ่นักศึกษาเขาใหมในแฟมนักศึกษา
R0103 : เพิม่ขอมูลการลงทะเบียนในแฟมลงทะเบียน
R0104 : ปรบัปรุงขอมูลการลงทะเบียนในแฟมลงทะเบียน
SUB02.PRG : ระบบงานหลังลงทะเบียน  ประกอบดวย 6 โปรแกรมยอย
R0201 : พมิพขอมูลการลงทะเบียนติดประกาศ
R0202 : พมิพรายชื่อนักเรียนใหอาจารยผูสอน
R0203 : พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
R0204 : พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช

R0205 : พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
R0206 : พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
SUB03.PRG : ระบบงานเมื่อถูกรองขอ  ประกอบดวย 4 โปรแกรมยอย
R0301 : ปรบัปรงุเกรดนักศึกษาที่ขอเพิกถอนวิชา
R0302 : พมิพตารางเรียนของนักศึกษาแตละวิชา
R0303 : พมิพใบรายงานผลการเรียนตลอดหลักสูตร
R0304 : ปรับปรุงแฟมอาจารย
SUB04.PRG : ระบบงานประจํ าภาคเรียน  ประกอบดวย 5 โปรแกรมยอย
R0401 : พมิพรายชื่อนักศึกษาแตละวิชาติดหนาหองสอบ
R0402 : พมิพรายชื่อแยกตามสถานภาพ
R0403 : พมิพรายชื่อแยกตามสาขา และจัดเรียงตามชื่อ
R0404 : ปรบัปรงุเกรดนักศึกษา เมื่ออาจารยสงเกรด
R0405 : พมิพใบรายงานผลการเรียนประจํ าภาค
SUB05.PRG : ระบบปรับปรุงและพิมพขอมูลจากแฟม  ประกอบดวย 5 โปรแกรมยอย
R0501 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมนักศึกษา
R0502 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมวิชา
R0503 : เพิ่ม ลบ แกไขหรือพิมพขอมูลจาก แฟมผูสอน
R0504 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมลงทะเบียน
R0505 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมหอง

5.3.4 วิธีการแปลโปรแกรม
การแปลโปรแกรม ในกรณีศึกษานี้ ใชคํ าสั่ง RMAKE ในการแปลโปรแกรม  โดยระบุชื่อโปรแกรมที่จะแปลไว

ในแฟม MAIN.RMK เพราะการแปลแบบนี้จะท ําใหการพัฒนาเร็วขึ้น  หากมีการเรียกโปรแกรมยอยจากภายนอกหลาย
ๆ โปรแกรม  เมื่อแกโปรแกรมยอยบางโปรแกรมจะตองแปลโปรแกรมทั้งหมดใหมดวยการแปลวิธีเดิม แตการแปลดวย
RMAKE จะเลอืกเฉพาะโปรแกรมยอยที่ถูกแกไขมารับการแปลใหมเทานั้น

จดุเดนของการใช  RMAKE.EXE  ชวยคือ การแปลโปรแกรมแตละครั้งจะเลือกเฉพาะโปรแกรมยอยที่ถูกปรับ
ปรงุมาแปลใหม  สวนที่ไมไดแกไขจะใชของแฟมเดิม  ท ําใหเร็วขึ้นอยางมากในการพัฒนาโปรแกรมที่มีการเรียกใช
โปรแกรมยอยจากภายนอกหลาย ๆ  โปรแกรม

การสรางแฟม MAIN.RMK
//  MAIN.RMK
//  ========
.PRG.OBJ:
      CLIPPER $< /M /N
.OBJ.EXE:
      SET RTLINKCMD=/POSI

       RTLINK $**, $@ ;
MAIN.OBJ:     MAIN.PRG
SUB01.OBJ:    SUB01.PRG
SUB02.OBJ:    SUB02.PRG
SUB03.OBJ:    SUB03.PRG
SUB04.OBJ:    SUB04.PRG
SUB05.OBJ:    SUB05.PRG
MAIN.EXE: MAIN.OBJ  SUB01.OBJ SUB02.OBJ SUB03.OBJ SUB04.OBJ SUB05.OBJ

การแปลโปรแกรมจะน ํา MAIN.RMK มารับการแปลดวย RMAKE โดยเขียนที่ DOS PROMPT ดังนี้
C:\CLIPPER5\RMAKE   MAIN      แตยงัคงใชแฟม CLIPPER.EXE และ RTLINK.EXE   เหมือนการแปลดวย
C:\CLIPPER5\CL MAIN

สิง่ทีแ่ตกตางของ RMAKE และ CL คือ RMAKE.EXE จะชวยเลือกโปรแกรมยอยที่มีการแกไขเทานั้นมา
ผานการแปลดวยโปรแกรมชุดเดิม แตไมไดน ําโปรแกรมยอยทั้งหมดมาแปลใหม

การเขยีนโปรแกรมให RMAKE แปล ตองเขียนทุกโปรแกรมใหอยูใน PROCEDURE หรือ FUNCTION  ถึงเปน
โปรแกรมหลักก็ตองมีโปรแกรมแรกอยูใน PROCEDURE แรกของโปรแกรม

&  5.4 โปรแกรมในระบบทะเบียน
โปรแกรมของระบบนี ้ไดจัดทํ าเปนโปรแกรมยอยไวภายนอก โดยมีโปรแกรม MAIN.PRG คอยเรียกโปรแกรม

เลก็ ๆ ที่อยูในโปรแกรมยอยอีกครั้งหนึ่ง  ส ําหรับโปรแกรมยอยภายนอกนั้น แตละโปรแกรมยอยคือระบบยอยระบบหนึ่ง
เพราะท ําใหงายตอการปรับปรุงแกไขและคนหา หากประมวลผลตามตัวเลือกแลวผิดพลาด

โปรแกรมยอยจากภายนอกมี 5 โปรแกรมคือ SUB01.PRG จนถึง SUB05.PRG  รวมโปรแกรมทั้งหมดใน
ระบบนี้เปน 6  โปรแกรม  โดยใหโปรแกรมยอยทั้ง 5 นี้ควบคุมระบบยอย 5 ระบบ

1. ระบบเตรียมงาน และงานลงทะเบียน
2. ระบบงานหลังลงทะเบียน
3. ระบบงานเมื่อถูกรองขอ
4. ระบบงานประจํ าภาคเรียน
5. ระบบปรับปรุงและพิมพขอมูลจากแฟม

5.4.1 โปรแกรมตาง ๆ ใน MAIN.PRG
โปรแกรมใน MAIN.PRG นี้มีโปรแกรมยอยทั้งหมด 21 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับเมนู และ

โปรแกรมสนับสนุน เชน โปรแกรมสอบถามปลายทางของผลลัพธ  โปรแกรมถามยํ้ าการปรับปรุงขอมูล โปรแกรมจัด
เรยีงขอมูล โปรแกรมแปลงเกรด โปรแกรมสั่งเลิกการท ํางาน เปนตน

โปรแกรมเมนูในโปรแกรม MAIN.PRG นี้มีหลักการงาย ๆ โดยสามารถเลือกตัวเลือกดวยคํ าสั่ง PROMPT และ
สงการท ํางานไปเมนูยอย ในขณะที่เมนูยอยแตละเมนูมีระบบการทํ างานที่สมบูรณในตนเอง จึงทํ าใหเมนูแบบนี้ทํ า
ความเขาใจไดงาย และน ําไปใชในทางปฎิบัติไดทันที

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม MAIN.PRG ทั้งหมด 21 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.1 ถงึ ตวัอยางที่ 5.14  (บางโปรแกรมสั้นและมีการท ํางานคลายกัน จึงนํ ามารวมในตัวอยางเดียวกัน)
: ตัวอยางที่ 5.1
&   ผลลัพธ
// โปรแกรมหลัก ของระบบงานทะเบียน เพื่อจัดทํ าเมนูหลัก แลวเลือกตัวเลือก
// เพือ่เขาไปสูเมนูยอย แตมีการระบุโปรแกรมยอยใหกับปุมฟงกชัน โดยระบุให
// ปุมฟงกชัน F1 เรียกโปรแกรมยอย HELPFORM เพื่อแสดงขอมูลผูเขียน
// ปุมฟงกชัน F2 เรียกโปรแกรมยอย HELPREADVAR เพื่อชวยผูใชปอนคาไดงาย
// ปุมฟงกชัน F10 เรียกโปรแกรมยอย QUITPROC เพื่อเลิกการทํ างาน
PROCEDURE MAINALL
   SET KEY 28 TO HELPFORM  // ควบคมุปุม F1 เพื่อแสดงขอมูลเกี่ยวกับผูเขียนโปรแกรม
   SET KEY -1 TO HELPREADVAR // ควบคมุปุม F2 เพื่อเลือกขอมูลไดอยางงาย ๆ
   SET KEY -9 TO QUITPROC    // ควบคมุปุม F10 เพื่อถามการเลิกทํ างาน
   SET WRAP ON
   SET DELIMITERS TO '[ ]'

   SET DELIMITERS ON
   SET EPOCH TO 1960
   DO WHILE .T.
      CLS ; OPT = 1
      @ 8,5 SAY "เมนูหลัก ระบบงานทะเบียน"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. ระบบเตรียมงานและงานลงทะเบียน"
      @ 11,5 PROMPT "2. ระบบงานหลังลงทะเบียน"
      @ 12,5 PROMPT "3. ระบบงานเมื่อถูกรองขอ"
      @ 13,5 PROMPT "4. ระบบงานประจํ าภาคเรียน"
      @ 14,5 PROMPT "5. ระบบปรับปรุงและพิมพขอมูลจากแฟม"
      @ 15,5 PROMPT "6. ขอมูลระบบ และทีมพัฒนาโปรแกรม"
      @ 16,5 PROMPT "7. ออกไปสูระบบปฏิบัติการชั่วคราว"
      @ 17,5 PROMPT "8. เลกิการท ํางาน"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <= 5 ;  DO &("SUB"+LTRIM(STR(OPT)))
         CASE OPT = 6 ;  DO WORKSYSTEM
         CASE OPT = 7
              IF MEMORY(2) > 400  ;  RUN COMMAND
              ELSE ; WAIT "หนวยความจํ าไมพอ จึงออกไประบบปฏิบัติการชั่วคราวไมได"
              ENDIF
         CASE OPT = 8 .OR. OPT = 0 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.2
&   ผลลัพธ
// สรางเมนยูอยส ําหรับระบบเตรียมงานและงานลงทะเบียน เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB1
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "1. เมนรูะบบเตรียมงานและงานลงทะเบียน"

      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. รบัขอมูลการเปดวิชาใหม"
      @ 11,5 PROMPT "2. รบัขอมูลขอมูลนักศึกษาใหม"
      @ 12,5 PROMPT "3. รบัลงทะเบียนจากนักศึกษา"
      @ 13,5 PROMPT "4. รบับอกเพิ่ม-ลดวิชา"
      @ 14,5 PROMPT "5. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=4 ;  DO &("R010"+LTRIM(STR(OPT)))
         CASE OPT = 5 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.3
&   ผลลัพธ
// สรางเมนูยอยส ําหรับระบบงานหลังลงทะเบียน เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB2
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "2. เมนูระบบงานหลังลงทะเบียน"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. พมิพขอมูลการลงทะเบียนติดประกาศ"
      @ 11,5 PROMPT "2. พมิพรายชื่อนักเรียนใหอาจารยผูสอน"
      @ 12,5 PROMPT "3. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร"
      @ 13,5 PROMPT "4. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช"
      @ 14,5 PROMPT "5. พมิพรายชื่อวิชาที่อาจารยผูสอนรับผิดชอบ"
      @ 15,5 PROMPT "6. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา"
      @ 16,5 PROMPT "7. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=6 ;  DO &("R020"+LTRIM(STR(OPT)))
         CASE OPT = 7 ;  EXIT

      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.4
&   ผลลัพธ
// สรางเมนูยอยส ําหรับระบบเมื่อถูกรองขอ เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB3
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "3. เมนูระบบเมื่อถูกรองขอ"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. การขอเพิกถอนวิชา"
      @ 11,5 PROMPT "2. ตรวจสอบตารางเรียนของนักศึกษาเฉพาะคน"
      @ 12,5 PROMPT "3. พมิพรายงานผลการเรียนตลอดหลักสูตร"
      @ 13,5 PROMPT "4. ปรบัปรุงขอมูลอาจารย"
      @ 14,5 PROMPT "5. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=4 ;  DO &("R030"+LTRIM(STR(OPT)))
         CASE OPT = 5 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.5
&   ผลลัพธ
// สรางเมนูยอยส ําหรับระบบงานประจ ําภาคเรียน เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB4
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "4. เมนูระบบงานประจ ําภาคเรียน"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. พมิพรายชื่อติดหนาหองสอบ"
      @ 11,5 PROMPT "2. พมิพรายชื่อแยกตามสถานภาพ"

      @ 12,5 PROMPT "3. พมิพรายชื่อแยกตามสาขา"
      @ 13,5 PROMPT "4. ใสเกรด"
      @ 14,5 PROMPT "5. พมิพรายงานผลการเรียนประจ ําภาค"
      @ 15,5 PROMPT "6. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=5 ;  DO &("R040"+LTRIM(STR(OPT)))
         CASE OPT = 6 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.6
&   ผลลัพธ
// สรางเมนูยอยสํ าหรับระบบปรับปรุง เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB5
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "5. เมนูระบบปรับปรุงและพิมพขอมูลจากแฟม"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. แฟมนักศึกษา"
      @ 11,5 PROMPT "2. แฟมวิชา"
      @ 12,5 PROMPT "3. แฟมผูสอน"
      @ 13,5 PROMPT "4. แฟมลงทะเบียน"
      @ 14,5 PROMPT "5. แฟมหอง"
      @ 15,5 PROMPT "6. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=5 ;  DO &("R050"+LTRIM(STR(OPT)))
         CASE OPT = 6 ;  EXIT
      ENDCASE
   ENDDO
RETURN

: ตัวอยางที่ 5.7
&   ผลลัพธ
// พมิพขอมูลของระบบทางจอภาพ เมื่อเลือกตัวเลือกที่ 6 จากเมนูหลัก
// โดยแสดงวันที ่เวลา ชื่อระบบปฏิบัติการ หนวยความจํ าที่เหลือ
// รุนทีใ่ชพัฒนาโปรแกรม สถานะของเครื่องพิมพ ตลอดจนขอมูลของ
// ผูพัฒนาโปรแกรม และสถานที่ติดตอ
PROCEDURE WORKSYSTEM
   CLS
   ? "                วนัที่ :",DTOC(DATE())
   ? "               เวลา :",TIME()
   ? "               วินาที :",SECOND()
   ? "     ระบบปฏิบัติการที่ใช :",OS()
   ? "ขนาดหนวยความจํ าที่เหลือ :",MEMORY(0)
   ? "          รุนของภาษา :",VERSION()
   ? " เครื่องพิมพเปดหรือปดอยู :",ISPRINTER()
   ? "====================================="
   ? "โปรแกรมนี้ถูกพัฒนาโดย. อ.บุรินทร รุจจนพันธุ"
   ? "เริ่มพัฒนาเมื่อ 8 กมุภาพันธ 2545"
   ? "โปรแกรมนี้เปนรุนที ่2.200202"
   ? "หากสนใจ หรือตองการตัวโปรแกรม ติดตอขอรับฟรีไดที"่
   ? "วทิยาลัยโยนก ล ําปาง 52000"
   ? "TEL.(054)265170 EXT.210   FAX. (054)265184"
   INKEY(2)
RETURN

: ตัวอยางที่ 5.8
&   ผลลัพธ
// ตวัอยางนี้ม ี2 ฟงกชันคือ ฟงกชันเพื่อเลือกปลายทางของผลลัพธ
// และฟงกชันถามยํ้ าการปรับปรุงขอมูล เมื่อมีการแกไขขอมูล
// ใชถามวาจะพิมพผลลัพธออกทางใด
FUNCTION ASKPRT()
   ASKPRT = ALERT("ผลลพัธของรายงาน",{'จอภาพ','เครื่องพิมพ'})
RETURN (ASKPRT)
// ใชถามวาจะยอมรับการปรับปรุงขอมูลหรือไม
FUNCTION ASKUPD()
   ASKUPD = ALERT("การปรบัปรุงขอมูล",{'ยอมรับการปรับปรุง','ยกเลิกการปรับปรุง'})
RETURN (ASKUPD)
: ตัวอยางที่ 5.9
&   ผลลัพธ
// เนือ่งจากระบบนี้ไมมีการใชแฟมดรรชน ีแตตองการการเรียงขอมูลใหถูกตอง
// จงึจดัท ําโปรแกรมยอย เพื่อเรียงขอมูลตามฟลดที่ตองการ โดยใชคํ าสั่งจัดเรียง
// ท ําใหผลการจัดเรียงถูกเก็บในแฟมใหม แลวจึงคัดลอกกลับมาทับแฟมเดิม
// ใชจดัเรียงแฟม REGIST ตามป ภาค รหัส วิชา
PROCEDURE SORTREG1
   USE REGIST
   SORT ON RYEAR,RSEM,IDSTD,SUBJ TO TEMPF
   USE TEMPF ; COPY TO REGIST ; CLOSE
RETURN
// ใชจดัเรียงแฟม REGIST ตามวิชา ป ภาค และรหัส
PROCEDURE SORTREG2
   USE REGIST
   SORT ON SUBJ,RYEAR,RSEM,IDSTD TO TEMPF
   USE TEMPF ; COPY TO REGIST ; CLOSE
RETURN
// ใชจดัเรียงแฟม SUBJECT ตามหองเรียน และชื่อวิชา
PROCEDURE SORTSUB1
   USE SUBJECT
   SORT ON ROOM,SUBJ TO  TEMPF
   USE TEMPF ; COPY TO SUBJECT ; CLOSE
RETURN

// ใชจดัเรียงแฟม SUBJECT ตามชื่อวิชา
PROCEDURE SORTSUB2
   USE SUBJECT
   SORT ON SUBJ TO  TEMPF
   USE TEMPF ; COPY TO SUBJECT ; CLOSE
RETURN
// ใชจัดเรียงแฟม IDSTD ตามชื่อรหัส
PROCEDURE SORTID1
   USE IDSTD
   SORT ON IDSTD TO TEMPF
   USE TEMPF ; COPY TO IDSTD ; CLOSE
RETURN
// ใชจดัเรียงแฟม IDSTD ตามชื่อสถานภาพ และรหัส
PROCEDURE SORTID2
   USE IDSTD
   SORT ON ISTAT,IDSTD TO  TEMPF
   USE TEMPF ; COPY TO IDSTD ; CLOSE
RETURN
// ใชจดัเรียงแฟม IDSTD ตามชื่อสาขา และชื่อ
PROCEDURE SORTID3
   USE IDSTD
   SORT ON IMAJOR,INAME,ISURN TO  TEMPF
   USE TEMPF ; COPY TO IDSTD ; CLOSE
RETURN
: ตัวอยางที่ 5.10
&   ผลลัพธ
// ฟงกชนัเปลี่ยนเกรด จากตัวเลข เปนตัวอักษร น ําไปแสดงผลไดชัดเจนขึ้น
FUNCTION  GRADECHR(RGRADE)
   GRADECHR = '--'
   DO CASE
      CASE RGRADE = 4    ;  GRADECHR = 'A '
      CASE RGRADE = 3.5  ;  GRADECHR = 'B+'
      CASE RGRADE = 3    ;  GRADECHR = 'B '
      CASE RGRADE = 2.5  ;  GRADECHR = 'C+'

      CASE RGRADE = 2    ;  GRADECHR = 'C '
      CASE RGRADE = 1.5  ;  GRADECHR = 'D+'
      CASE RGRADE = 1    ;  GRADECHR = 'D '
      CASE RGRADE = 0    ;  GRADECHR = 'F '
      CASE RGRADE = 4.2  ;  GRADECHR = 'W '
      CASE RGRADE = 4.6  ;  GRADECHR = 'P '
   ENDCASE
RETURN (GRADECHR)
: ตัวอยางที่ 5.11
&   ผลลัพธ
// ฟงกชนัปลี่ยนเกรด จากตัวอักษรเปนตัวเลข ท ําใหจัดเก็บลงแฟมไดงายขึ้น
FUNCTION  GRADENUM(_GRADE)
   GRADENUM = 0
   DO CASE
      CASE UPPER(_GRADE) = 'A'  ; GRADENUM = 4
      CASE UPPER(_GRADE) = 'B+' ; GRADENUM = 3.5
      CASE UPPER(_GRADE) = 'B'  ; GRADENUM = 3
      CASE UPPER(_GRADE) = 'C+' ; GRADENUM = 2.5
      CASE UPPER(_GRADE) = 'C'  ; GRADENUM = 2
      CASE UPPER(_GRADE) = 'D+' ; GRADENUM = 1.5
      CASE UPPER(_GRADE) = 'D'  ; GRADENUM = 1
      CASE UPPER(_GRADE) = 'F'  ; GRADENUM = 0
      CASE UPPER(_GRADE) = 'W'  ; GRADENUM = 4.2
      CASE UPPER(_GRADE) = 'P'  ; GRADENUM = 4.6
   ENDCASE
RETURN (GRADENUM)
: ตัวอยางที่ 5.12
&   ผลลัพธ
// แสดงขอความงาย ๆ ทันท ีเมื่อกดปุมฟงกชัน F1 โดยไมท ําใหภาพเดิมบนจอเปลี่ยน
// เพราะมกีารใชค ําสั่งเก็บสถานะของจอ และคืนสถานะเดิมของจอ และที่ใช F1 ได
// เพราะ ระบุในโปรแกรมหลัก MAINALL วา SET KEY 28 TO HELPFORM
// แสดงขอมูลของผูพัฒนาโปรแกรมใหทราบ

PROCEDURE HELPFORM
   SAVE SCREEN TO SCR
   @ 5,5 CLEAR TO 20,75 // ลบขอความบนพื้นที่กํ าหนด
   @ 9,8 TO 15,71  // ตกีรอบรอบขอความ
   @ 10,10 SAY "CLIPPER PROGRAMING"
   @ 11,10 SAY "LEARING BY CASE STUDY"
   @ 12,10 SAY "GOOD LUCK"
   @ 13,10 SAY DATE()
   @ 14,10 SAY TIME()
   INKEY(5)
   RESTORE SCREEN FROM SCR
RETURN
: ตัวอยางที่ 5.13
&   ผลลัพธ
// ฟงกชนันีช้วยใหการใสคาในตัวแปรบางตัวแปรถูกตอง เพราะจะแสดง
// คาทีเ่หมาะสมกับฟลดนั้นมาใหเลือก โดยการอานคาจากแฟมมาแสดงให
// ด ูแตจะตองระบุใหชัดเจนวาจะน ําฟลดใดมาใชบาง
// ฟงกชันนี้จึงแสดงขอมูลที่เปนไปไดส ําหรับตัวแปรที่กํ าลังรอรับคา
FUNCTION HELPREADVAR
   OLDALIAS = ALIAS() // เก็บพื้นที่ท ํางานเดิม
   SHOWAR = {}
   DO CASE
      CASE READVAR() = "_SUBJ"
           IF OLDALIAS != "SUBJECT"
              USE SUBJECT NEW // ถาเปดแลวตองไมเปดอีก
           ENDIF
           GO TOP
           WHILE !EOF()
              AADD(SHOWAR,SUBJ)
              SKIP
           END
           IND := ACHOICE(2,71,5,78,SHOWAR)
           IF IND > 0 // ปองกันการใชปุม ESC
              _SUBJ = SHOWAR[IND]

           ENDIF
           IF OLDALIAS != "SUBJECT" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
              CLOSE                 // ในจุดที่ไมไดเปดแฟมนี้
           ENDIF
      CASE READVAR() = "_IDSTD"
           IF OLDALIAS != "IDSTD"
              USE IDSTD NEW  // ถาเปดแลวตองไมเปดอีก
           ENDIF
           GO TOP
           FOR I = 1 TO  RECCOUNT()
               AADD(SHOWAR,STR(I,2)+". "+STR(IDSTD)+" : "+INAME)
               SKIP
           NEXT
           IND := ACHOICE(2,50,10,78,SHOWAR)
           IF IND > 0  // ปองกันการใชปุม ESC
              _IDSTD = VAL(SUBSTR(SHOWAR[IND],5,7))
           ENDIF
           IF OLDALIAS != "IDSTD" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
              CLOSE               // ในจุดที่ไมไดเปดแฟมนี้
           ENDIF
       OTHERWISE
           X = ALERT("ขอโทษ ไมไดเตรียมสวนสนับสนุน ส ําหรับจุดนี้",{"OK."})
   ENDCASE
   IF LEN(OLDALIAS) > 0
      SELE &OLDALIAS // กลับสูพื้นที่ท ํางานเดิม
   ENDIF
RETURN

: ตัวอยางที่ 5.14
&   ผลลัพธ
// สอบถามการเลิกท ํางานดวยฟงกชัน ALERT
PROCEDURE QUITPROC
   Q =  ALERT("สอบถามการเลิกการท ํางาน",{"เลิกการทํ างาน","ยังไมเลิกงาน"})
   IF Q = 1
      QUIT
   ENDIF
RETURN

5.4.2 โปรแกรมตาง ๆ ใน SUB01.PRG
โปรแกรมใน SUB01.PRG นี้มีโปรแกรมยอยทั้งหมด 4 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบเตรียม

งาน และงานลงทะเบียน เชน โปรแกรมรับขอมูลการลงทะเบียน โปรแกรมขอบอกเพิ่มลดวิชา โปรแกรมปอนขอมูลนัก
ศกึษา หรือโปรแกรมปรับปรุงรายวิชาเปนตน

ระบบเตรยีมงาน และงานลงทะเบียนเปนระบบที่ส ําคัญยิ่ง เพราะระบบนี้ถูกพัฒนาขึ้นเพื่อเก็บประวัติการ
เรยีน และโปรแกรมในระบบนี้จะบันทึกขอมูลการลงทะเบียนของนักศึกษาแตละคน  ท ําใหเกิดขอมูลขึ้น เสมือนวาจุดนี้
เปนจดุเริ่มตนที่ท ําใหเกิดขอมูล  น ําไปปรับปรุง น ําไปใช และประมวลผลในงานของระบบอื่นไดอีกมาก

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB01.PRG ทั้งหมด 4 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.15 ถึง ตัวอยางที่ 5.18 ( R0101 - R0104 )
: ตัวอยางที่ 5.15
&   ผลลัพธ
// ปรบัปรงุขอมูลในแฟมวิชาอยางงาย ๆ ดวยฟงกชัน BROWSE
// แลวสอบถามวาจะปรับปรุงหรือไม ถาไมจะยกเลิกการปรับปรุงทั้งหมด
PROCEDURE R0101
   USE SUBJECT
   COPY TO TEMPF
   BROWSE(5,5,22,75)
   IF ASKUPD() = 1
      PACK
   ELSE
      USE TEMPF
      COPY TO SUBJECT
   ENDIF
   CLOSE ALL
RETURN

: ตัวอยางที่ 5.16
&   ผลลัพธ
// เพิม่รายชือ่นักศึกษาใหม เพียงคนเดียว โดยระบุรหัสใหมใหอัตโนมัติ
PROCEDURE R0102
   USE IDSTD
   GOTO BOTTOM
   _IDSTD = IDSTD + 1
   _INAME = SPACE(20)
   _ISURN = SPACE(20)
   _IMAJOR = 0
   _ISTAT = 'A'
   @ 10,5 SAY "เพิม่รายชื่อนักศึกษาใหม"
   @ 11,5 SAY "รหัส"+STR(_IDSTD)
   @ 12,5 SAY "ชื่อ" GET _INAME
   @ 13,5 SAY "สกุล" GET _ISURN
   @ 14,5 SAY "สาขา" GET _IMAJOR
   @ 15,5 SAY "สถานภาพ" GET _ISTAT
   READ
   IF ASKUPD() = 1 // สอบถามการปรับปรุง
      APPEND BLANK
      REPLACE IDSTD WITH _IDSTD ,INAME WITH _INAME ,ISURN WITH _ISURN,;
              IMAJOR WITH _IMAJOR,ISTAT WITH _ISTAT
   ENDIF
RETURN
: ตัวอยางที่ 5.17
&   ผลลัพธ
// เพิม่ขอมูลการลงทะเบียนของนักศึกษาแตละคน
// วธินีีจ้ะลงทะเบียนและตัดยอดใหทันท ีส ําหรับแตละวิชา
// หากลง 5 วิชาแตผิดพลาด 2 วิชา จะเพิ่มขอมูลเขาไป 3 วิชาทันที
// การปรบัปรงุจะเพิ่มจ ํานวนอีก 1 เขาไปในฟลดจํ านวนที่ลง ในแฟมวิชาแตละวิชา
// และเพิม่เรคอรดเทาจ ํานวนวิชาที่ลงทะเบียนไดในแฟมลงทะเบียน
// พมิพใบเสร็จโดยคิดคาลงทะเบียนเรียนวิชาละ 1000 บาท
PROCEDURE R0103
   SET CENTURY ON

   _RYEAR = IIF(MONTH(DATE())<5,YEAR(DATE())-1,YEAR(DATE()))
   _RSEM = '1'
   _IDSTD = 0
   _SUBJ := ARRAY(8)
   _RGRADE = 4.6
   @ 4,5 SAY "โปรแกรม ลงทะเบียนเรียน"
   @ 5,5 SAY "ประจ ําปการศึกษา : " GET _RYEAR PICT '9999'
   @ 6,5 SAY "   ภาคการศึกษา : " GET _RSEM
   SAVE SCREEN TO SCR
   DO WHILE LASTKEY() != 27
      AFILL(_SUBJ,SPACE(8))  // ก ําหนดคาเริ่มตนใหตัวแปรอาเรย _SUBJ ทุกสมาชิก
      _IDSTD = 0
      DO WHILE .T. // รบัขอมูลนักศึกษาที่มีในแฟม IDSTD เทานั้น
         USE IDSTD
         @ 7,5 SAY "    รหสันักศึกษา : " GET _IDSTD PICT '9999999'
         READ
         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            _INAME = INAME
            _ISURN = ISURN
            @ 8,5 SAY INAME+' '+ISURN+' '+STR(IMAJOR)+' '+ISTAT
            EXIT
         ELSE ; @ 8,5 SAY 'ไมมีรหัสนักศึกษานี้ในแฟม IDSTD'
         ENDIF
      ENDDO
      @ 10,10 SAY "SUBJ1 : " GET _SUBJ[1] PICT '@!'
      @ 11,10 SAY "SUBJ2 : " GET _SUBJ[2] PICT '@!'
      @ 12,10 SAY "SUBJ3 : " GET _SUBJ[3] PICT '@!'
      @ 13,10 SAY "SUBJ4 : " GET _SUBJ[4] PICT '@!'
      @ 14,10 SAY "SUBJ5 : " GET _SUBJ[5] PICT '@!'
      @ 15,10 SAY "SUBJ6 : " GET _SUBJ[6] PICT '@!'
      @ 16,10 SAY "SUBJ7 : " GET _SUBJ[7] PICT '@!'
      @ 17,10 SAY "SUBJ8 : " GET _SUBJ[8] PICT '@!'

      READ
      YOURSUBJ = 8  // จ ําเปนตอง 8 เพราะตรวจชองวางของวิชา
      OKSUBJ = 0
      USE SUBJECT
      FOR I = 1 TO 8 // น ําวชิามาตรวจสอบในแฟมวิชา แลวเพิ่มจํ านวนเขาไป
         IF _SUBJ[I] != SPACE(8)
            LOCATE FOR _SUBJ[I] = SUBJ
            IF FOUND()
               IF SREGIST >= SLIMIT
                  @ 9+I,40 SAY 'วชิานีเ้ตม็แลว เพราะรับเพียง'+STR(SLIMIT)
               ELSE
                  @ 9+I,40 SAY 'OK.' // วชิาลงเพิ่ม จึงเพิ่มคาในฟลดจํ านวนที่ลง
                  REPLACE SREGIST WITH SREGIST + 1
                  USE REGIST          // เพิม่เรคอรดเขาไปในแฟม REGIST
                     APPEND BLANK       // 4.6 คอืเกรด P หมายถึงกํ าลังศึกษาอยู
                     REPLACE RYEAR WITH _RYEAR, RSEM WITH _RSEM ,;
                        IDSTD WITH _IDSTD, SUBJ WITH _SUBJ[I],RGRADE WITH 4.6
                  USE SUBJECT
                  OKSUBJ = OKSUBJ + 1
               ENDIF
            ELSE
               @ 9+I,40 SAY 'ไมมีวิชานี้ในแฟม SUBJECT'
            ENDIF
         ELSE
            YOURSUBJ = I - 1
            I = 8
         ENDIF
      NEXT
      @ 18,10 SAY 'จ ํานวนวชิาที่ทานลงทะเบียนเรียนคือ '+STR(YOURSUBJ)
      @ 19,10 SAY 'จ ํานวนวิชาที่ทานลงไดคือ '+STR(OKSUBJ)
      INKEY(2)
      // สวนนี้ใชพิมพใบเสร็จ โดยคิดวิชาละ 1000 บาท
      IF OKSUBJ >= 1

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                            บทที่ 5  :  กรณศีกึษาระบบงานทะเบียน                                                      251

         GASKPRT = ASKPRT()
         IF GASKPRT = 2 .AND. ISPRINTER()  // เลข 2 หมายถึงเครื่องพิมพ
            SET DEVICE TO PRINTER
         ELSE ; CLS
         ENDIF
         @ 1,5  SAY "รหัส :"+STR(_IDSTD)
         @ 2,5  SAY "ชื่อ-สกุล :"+_INAME+" "+_ISURN
         @ 3,5  SAY "==========================="
         FOR I = 1 TO YOURSUBJ
            @ 3+I,5 SAY _SUBJ[I]
         NEXT
         @ 4+YOURSUBJ,5 SAY "จ ํานวนเงนิทั้งหมดที่ทานตองช ําระ"+STR(OKSUBJ*1000)
         IF GASKPRT = 2 .AND. ISPRINTER() ; EJECT ; ENDIF
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(0) // ขณะรอกดปุม ถากด ESC จะเลิกงานลงทะเบียน ถาไมจะรับคนใหม
      RESTORE SCREEN FROM SCR
   ENDDO
   SET CENTURY OFF
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.18
&   ผลลัพธ
// บอกเพิ่ม-ลดวิชา เพื่อปรับปรุงขอมูลการลงทะเบียน
// การปรบัปรงุส ําหรับการบอกเพิ่มคือเพิ่มจํ านวนในฟลดจํ านวนที่ลง แตบอกเลิกจะลดลง
// และจะเพิ่มหรือลดเรคอรดเทาจ ํานวนที่ระบุในแฟมลงทะเบียน
// สดุทายพมิพใบเสร็จ โดยคิดวิชาละ 1000 บาท และคาธรรมเนียมวิชาละ 20 บาท
PROCEDURE R0104
   SET CENTURY ON
   _RYEAR = IIF(MONTH(DATE())<5,YEAR(DATE())-1,YEAR(DATE()))
   _RSEM = '1'
   _IDSTD = 0
   _SUBJ := ARRAY(8)
   @ 4,5 SAY "โปรแกรม บอกเพิ่ม-ลดวิชา"

   @ 5,5 SAY "ประจํ าปการศึกษา : " GET _RYEAR
   @ 6,5 SAY "   ภาคการศึกษา : " GET _RSEM
   SAVE SCREEN TO SCR
   DO WHILE LASTKEY() != 27
      AFILL(_SUBJ,SPACE(8))  // ก ําหนดคาเริ่มตนใหตัวแปรอาเรย _SUBJ ทุกสมาชิก
      _IDSTD = 0
      @ 7,5 SAY "    รหสันักศึกษา : " GET _IDSTD PICT '9999999'
      READ
      @ 8,10 SAY "เพิ่มวิชา"
      @ 9,10 SAY "SUBJ1 : " GET _SUBJ[1] PICT '@!'
      @ 10,10 SAY "SUBJ2 : " GET _SUBJ[2] PICT '@!'
      @ 11,10 SAY "SUBJ3 : " GET _SUBJ[3] PICT '@!'
      @ 12,10 SAY "SUBJ4 : " GET _SUBJ[4] PICT '@!'
      @ 13,10 SAY "ลดวิชา"
      @ 14,10 SAY "SUBJ5 : " GET _SUBJ[5] PICT '@!'
      @ 15,10 SAY "SUBJ6 : " GET _SUBJ[6] PICT '@!'
      @ 16,10 SAY "SUBJ7 : " GET _SUBJ[7] PICT '@!'
      @ 17,10 SAY "SUBJ8 : " GET _SUBJ[8] PICT '@!'
      READ
      YOURADD := YOURDROP := 4
      OKSUBJADD := OKSUBJDROP := 0
      USE SUBJECT
      FOR I = 1 TO 4
         IF _SUBJ[I] != SPACE(8)
            LOCATE FOR _SUBJ[I] = SUBJ
            IF FOUND()
               IF SREGIST >= SLIMIT
                  @ 8+I,40 SAY 'วชิานีเ้ตม็แลว เพราะรับเพียง'+STR(SLIMIT)
               ELSE
                  @ 8+I,40 SAY 'OK.' // วชิาลงเพิ่ม จึงเพิ่มคาในฟลดจํ านวนที่ลง
                  REPLACE SREGIST WITH SREGIST + 1
                  USE REGIST
                     APPEND BLANK      // 4.6 คอืเกรด P หมายถึงกํ าลังศึกษาอยู

                     REPLACE RYEAR WITH _RYEAR, RSEM WITH _RSEM ,;
                        IDSTD WITH _IDSTD, SUBJ WITH _SUBJ[I],RGRADE WITH 4.6
                  USE SUBJECT
                  OKSUBJADD = OKSUBJADD + 1
               ENDIF
            ENDIF
         ELSE
            YOURADD = I - 1   ; I = 4
         ENDIF
      NEXT
      USE REGIST  // การบอกเลิกจะตองมีเรคอรดที่นั้นกอนจึงจะบอกเลิกได
      FOR I = 5 TO 8
         IF _SUBJ[I] != SPACE(8)
            LOCATE FOR _RYEAR = RYEAR .AND. _RSEM = RSEM .AND. ;
                       _IDSTD = IDSTD .AND. _SUBJ[I] = SUBJ
            IF FOUND()
               DELETE
               USE SUBJECT
               LOCATE FOR _SUBJ[I] = SUBJ
               IF FOUND()
                  @ 9+I,40 SAY 'OK.' // วชิานีม้คีนบอกเลิก จึงลดคาใน SREGIST
                  REPLACE SREGIST WITH SREGIST - 1
                  OKSUBJDROP = OKSUBJDROP + 1
               ENDIF
               USE REGIST
            ENDIF
         ELSE
            YOURDROP = I - 5   ; I = 8
         ENDIF
      NEXT
      PACK  // ลบเรคอรดที่ท ําเครื่องหมายลบไว ออกจากแฟมอยางแทจริง
      @ 18,10 SAY 'จ ํานวนวชิาที่ทานบอกเพิ่ม '+STR(YOURADD)
      @ 19,10 SAY 'จ ํานวนวชิาที่ทานบอกเลิก '+STR(YOURDROP)

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


254                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

      @ 20,10 SAY 'จ ํานวนวิชาที่บอกเพิ่มได '+STR(OKSUBJADD)
      @ 21,10 SAY 'จ ํานวนวชิาที่บอกเลิกได '+STR(OKSUBJDROP)
      INKEY(2)
      // พมิพใบเสร็จ คิดวิชาละ 1000 บาท และคาธรรมเนียมเพิ่มลด วิชาละ 20 บาท
      IF (OKSUBJADD + OKSUBJDROP) >= 1
         GASKPRT = ASKPRT()
         IF GASKPRT = 2 .AND. ISPRINTER()  // เลข 2 หมายถึงเครื่องพิมพ
            SET DEVICE TO PRINTER
         ELSE ; CLS
         ENDIF
         @ 1,5  SAY "รหัส :"+STR(_IDSTD)
         @ 2,5  SAY "==========================="
         FOR I = 1 TO YOURADD  // พมิพวิชาที่บอกเพิ่ม
            @ 3+I,5 SAY 'A : '+ _SUBJ[I]
         NEXT
         FOR I = 1 TO YOURDROP // พมิพวิชาที่บอกเลิก
            @ 3+I+YOURADD,5 SAY 'D : '+ _SUBJ[I+4]
         NEXT
         @ 4+YOURADD+YOURDROP,5 SAY "จ ํานวนเงินทั้งหมดที่ทานตองชํ าระ หรือรับคืน"+ ;
               STR(((OKSUBJADD-OKSUBJDROP)*1000)+(OKSUBJADD+OKSUBJDROP)*20)
         IF GASKPRT = 2 .AND. ISPRINTER() ; EJECT ; ENDIF
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(0) // ขณะรอกดปุม ถากด ESC จะเลิกงานลงทะเบียน ถาไมจะรับคนใหม
      RESTORE SCREEN FROM SCR
   ENDDO
   SET CENTURY OFF
RETURN

5.4.3 โปรแกรมตาง ๆ ใน SUB02.PRG
โปรแกรมใน SUB02.PRG นี้มีโปรแกรมยอยทั้งหมด 6 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบงาน หลัง

ลงทะเบยีน เชน โปรแกรมพิมพรายชื่อนักศึกษาที่ลงทะเบียนติดประกาศ  พิมพรายชื่อใหอาจารยผูสอน  พิมพรายชื่อ
วชิาตดิหนาหองเรียน  พิมพสรุปรายงานใหผูบริหาร  เปนตน

ระบบงานหลังลงทะเบียนเปนระบบที่ท ําใหเกิดประสิทธิภาพในการเรียนการสอน เพราะอาจารยผูสอนจะได
รบัรายชือ่ เพือ่ตรวจสอบเวลาเรียนและลงคะแนน ผูบริหารทราบจ ํานวนนักศึกษาที่ลงทะเบียน ทราบสถิติการใชหอง
เรยีนวามปีระสทิธิภาพเพียงใด  ดังนั้นระบบนี้จึงมุงที่จะทํ ารายงานใหหนวยงานตาง ๆ หลังลงทะเบียนทันที

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB02.PRG ทั้งหมด 6 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.19 ถึง ตัวอยางที่ 5.24 ( R0201 - R0206 )
: ตัวอยางที่ 5.19
&   ผลลัพธ
// พมิพขอมูลการลงทะเบียนเรียนติดประกาศ ใหนักศึกษาดู
// รายงานแสดงล ําดับ รหัสนักศึกษา และรหัสวิชา
// เมือ่เริม่โปรแกรมจะท ําการจัดเรียงขอมูลใหมหมด แลวแยกกลุมขอมูล โดยใชรหัส
// จะน ําขอมูลเฉพาะป 1996 ภาคเรียนที่ 1 เทานั้นมาแสดง
// ในขณะพมิพขอมูลสามารถนับจ ํานวนของนักศึกษา และวิชาของนักศึกษาแตละคน
PROCEDURE R0201
   DO SORTREG1 // ใชจดัเรียงขอมูลตามป ภาค รหัส และวิชา อยูใน MAIN.PRG
   USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
   GO TOP
   _IDSTDOLD = 0
   IIDSTD := ISUBJ := 0
   DO WHILE !EOF()
      IF IDSTD <> _IDSTDOLD
        IIDSTD++
        ISUBJ = 0
        INKEY(0.5)
        ? '* --> ',IIDSTD,IDSTD
        _IDSTDOLD = IDSTD
      ENDIF
      ISUBJ++
      ? ISUBJ,SUBJ
      SKIP
   ENDDO
   INKEY(0)
   CLOSE ALL
RETURN

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


256                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

: ตัวอยางที่ 5.20
&   ผลลัพธ
// พมิพรายชื่อนักศึกษาใหอาจารยผูสอน
// รายงานแสดงล ําดับ รหัสวิชา รหัสนักศึกษา ชื่อ สกุล รหัสสาขา และสถานภาพ
// เมือ่เริม่โปรแกรมตองจัดเรียงขอมูลใหเรียงตามรหัสวิชากอน
PROCEDURE R0202
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG
   SELE 1
   USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
   GO TOP
   SELE 2
   USE IDSTD
   SELE 1
   _SUBJOLD = SPACE(8)
   IIDSTD := ISUBJ := 0
   DO WHILE !EOF()
      IF SUBJ <> _SUBJOLD
         ISUBJ++
         IIDSTD = 0
         INKEY(0.5)
         ? '* --> ',ISUBJ,SUBJ
         _SUBJOLD = SUBJ
      ENDIF
      IIDSTD++
      _IDSTD = IDSTD
      SELE 2
      LOCATE FOR IDSTD = _IDSTD
      IF FOUND()
         ? IIDSTD,IDSTD,INAME,ISURN,IMAJOR,ISTAT
      ELSE ; ? IIDSTD,_IDSTD  // เพราะรหัสนี้ไมมีใน IDSTD จึงตองใชตัวแปรแทน
      ENDIF
      SELE 1 // ตองมาที่พื้นที ่1 จึงจะ SKIP เพราะตองการเลื่อนเรคอรดใน REGIST
      SKIP

   ENDDO
   INKEY(0)
   CLOSE ALL // ตอง CLOSE ALL เพราะเปดแฟมไวหลายพื้นที่
RETURN
: ตัวอยางที่ 5.21
&   ผลลัพธ
// พมิพสรุปการลงทะเบียนใหผูบริหาร โดยพิมพจํ านวนที่ลงเรียนในแตละวิชา
// โปรแกรมนี้คุมกลุม 2 แบบคือ กลุมของวิชา และกลุมของภาคการศึกษา
// รายงานแสดงปการศึกษา ภาคการศึกษา วิชา จํ านวนที่ลงทะเบียนเรียน
// สามารเลือกปลายทางของผลลัพธได
// ควบคมุจ ํานวนบรรทัดตอหนาของจอภาพเปน 20 และของเครื่องพิมพเปน 30
PROCEDURE R0203
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG
   USE REGIST
   _SUBJO = SUBJ
   _RYEARO = RYEAR
   _RSEMO = RSEM
   IIDSTD := ISUBJ := 0
   GASKPRT = ASKPRT()
   IF GASKPRT = 2 // เลข 2 หมายถึงพิมพรายงานทางเครื่องพิมพ
      SET DEVICE TO PRINTER
      CNTPAGE = 30
   ELSE
      CNTPAGE = 20
   ENDIF
   CNTLINE = 1
   DO WHILE !EOF()
      IIDSTD++
      IF SUBJ <> _SUBJO .OR. ISUBJ = 0   // ตรวจสอบความตางของวิชา
         INKEY(0.5)
         ISUBJ++
         @ CNTLINE,5 SAY STR(ISUBJ)+SUBJ
         _SUBJO = SUBJ
         CNTLINE++

      ENDIF
      SKIP           // ตรวจสอบภาคการศึกษา
      IF SUBJ <> _SUBJO .OR. _RYEARO <> RYEAR .OR. _RSEMO <> RSEM .OR. EOF()
         @ CNTLINE,10 SAY STR(_RYEARO) + _RSEMO + STR(IIDSTD)
         CNTLINE++
         _RYEARO = RYEAR
         _RSEMO = RSEM
         IIDSTD = 0
      ENDIF
      IF CNTLINE >= CNTPAGE .OR. EOF() // ตรวจสอบการแบงหนา
         CNTLINE = 1
         IF GASKPRT = 2 ; EJECT
         ELSE ; INKEY(3) ; CLS ; ENDIF
      ENDIF
   ENDDO
   SET DEVICE TO SCREEN
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.22
&   ผลลัพธ
// พมิพรายชื่อวิชาของแตละหอง โดยเลือกปลายทางของผลลัพธได
// รายงานแสดงล ําดับและรหัสหอง รหัสวิชา เวลาเรียน และจ ํานวนที่ลงทะเบียน
// โปรแกรมนี้จะแบงหองเรียน แลวรายงานชื่อวิชาในหองนั้น
// โดยรายงานจะแสดงหองละ 1 หนาเทานั้น
PROCEDURE R0204
   DO SORTSUB1   // ใชจดัเรยีงแฟมวิชาตามหองเรียนและชื่อวิชา
   USE SUBJECT
   _ROOMO = SPACE(4)
   IROOM := ISUBJ := 0
   GASKPRT = ASKPRT()
   IF GASKPRT = 2
      SET DEVICE  TO PRINTER
   ENDIF

   DO WHILE !EOF()
      IF ROOM <> _ROOMO
         IF GASKPRT = 2 .AND. IROOM != 0
            EJECT
         ELSE
            INKEY(3)
            CLS
         ENDIF
         IROOM++
         @ 1,1 SAY "รายวชิาที่มีการใชในแตละหอง"
         @ 2,1 SAY "หองที ่"+STR(IROOM)+" เบอร "+ROOM
         @ 3,1 SAY "============================"
         _ROOMO = ROOM
         ISUBJ = 0
      ENDIF
      ISUBJ++
      @ ISUBJ+3,1 SAY STR(ISUBJ) + SUBJ + STIME + STR(SREGIST)
      SKIP
   ENDDO
   SET DEVICE TO SCREEN
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.23
&   ผลลัพธ
// พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
// รายงานแสดงรหัสอาจารย ชื่ออาจารย รหัสวิชา หองเรียน และเวลาเรียน
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE R0205
   DO SORTSUB2
   SELE 1
   USE TEACHER
   SELE 2
   USE SUBJECT

   IF ASKPRT() = 2  // ไดผลลัพธออกเครื่องพิมพ
      SET DEVICE TO PRINTER
      FOR I= 1 TO RECCOUNT()
          @ I,10 SAY STR(I) + SUBJ + ' | ' + ROOM + ' | ' + STIME
          SELE 1
             LOCATE FOR SUBJECT->TEACHID = TEACHID
             IF FOUND()
                @ I,50 SAY ' | ' + TNAME + ' | ' + TSURN
             ENDIF
             IF MOD(I,15) = 0
                EJECT
             ENDIF
          SELE 2
          SKIP
      NEXT
      EJECT
      SET DEVICE TO SCREEN
   ELSE
      I = 0  // ไดผลลัพธออกทางจอภาพ
      DO WHILE !EOF()
         I++
         ? I,SUBJ,ROOM,STIME
         SELE 1
            LOCATE FOR SUBJECT->TEACHID = TEACHID
            IF FOUND() ; ?? TNAME,TSURN ; ENDIF
            IF I = 15
               INKEY(3)
               CLS
               I = 0
            ENDIF
         SELE 2
         SKIP
      ENDDO

   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.24
&   ผลลัพธ
// พมิพสรุปจ ํานวนนักศึกษาแตละวิชา โดยระบุป และภาค
// รายงานแสดงล ําดับ รหัสวิชา จํ านวนที่ลงทะเบียน
// สามารถเลือกรูปแบบของผลลัพธได
PROCEDURE R0206
   _RYEAR = 0
   _RSEM = '0'
   @ 4,5 SAY 'สรปุจ ํานวนนักศึกษาในปและภาคที่ระบุ'
   @ 5,5 SAY 'ปการศึกษา : ' GET _RYEAR
   @ 6,5 SAY 'ภาคการศึกษา : ' GET _RSEM
   READ
   CLS
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG
   USE REGIST
   SET FILTER TO RYEAR = _RYEAR .AND. RSEM = _RSEM
   GO TOP
   _SUBJO = SUBJ
   IIDSTD := ISUBJ := 0
   GASKPRT = ASKPRT()
   CNTLINE = 1
   CNTPAGE = 20 // เริ่มตนส ําหรับจอภาพใหหนาละ 20 บรรทัด
   IF GASKPRT = 2 // เลข 2 หมายถึงพิมพรายงานทางเครื่องพิมพ
      SET DEVICE TO PRINTER
      CNTPAGE = 30
   ENDIF
   @ CNTLINE,5 SAY "สรปุจ ํานวนนักศึกษาประจํ าป " + STR(_RYEAR) + "/" + _RSEM
   CNTLINE++
   DO WHILE !EOF()
      IIDSTD++

      SKIP
      IF SUBJ <> _SUBJO .OR. EOF()   // ตรวจสอบความตางกลุมของวิชา
         ISUBJ++
         @ CNTLINE,5 SAY STR(ISUBJ)+_SUBJO+STR(IIDSTD)
         _SUBJO = SUBJ
         CNTLINE++
         IIDSTD = 0
      ENDIF
      IF CNTLINE >= CNTPAGE .OR. EOF()  // ตรวจสอบการแบงหนา
         CNTLINE = 1
         IF GASKPRT = 2 ; EJECT
         ELSE ; INKEY(3) ; CLS ; ENDIF
      ENDIF
   ENDDO
   SET DEVICE TO SCREEN
   INKEY(3)
   CLOSE ALL
RETURN

5.4.4 โปรแกรมตาง ๆ ใน SUB03.PRG
โปรแกรมใน SUB03.PRG นี้มีโปรแกรมยอยทั้งหมด 4 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบงาน เมื่อ

ถกูรองขอ เชน โปรแกรมขอถอนวิชาเรียน  โปรแกรมพิมพตารางเรียน โปรแกรมพิมพใบรายงานผลการเรียนโปรแกรม
ปรบัปรุงขอมูลอาจารย เปนตน

ระบบงานเมือ่ถูกรองขอ เปนระบบเกิดขึ้นเมื่อมีผูรองขอปรับปรุง หรือรายงานจากระบบ โดยไมขึ้นกับเวลา วา
จะเปนตนภาคเรียน ทายภาค หรือชวงปดภาคเรียน  แตในระบบอื่น ๆ ผูจัดระบบอาจแยกโปรแกรมตาง ๆ ในระบบนี้เขา
ไปรวมกบัระบบอื่นได  แตผูเขียนแยกออกมาเพื่อใหเห็นการแยกระบบยอย ในอีกมุมมองหนึ่ง

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB03.PRG ทั้งหมด 4 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.25 ถึง ตัวอยางที่ 5.28 ( R0301 - R0304 )
: ตัวอยางที่ 5.25
&   ผลลัพธ
// ปอนขอมลูการขอถอนวิชาของนักศึกษา (W) โดยระบุรหัสนักศึกษา และวิชา
// ส ําหรบัปการศึกษาและภาคเรียนจะระบุไวแนนอนในโปรแกรมเปนป 1996 ภาคเรียนที่ 1
// โปรแกรมจะคนหาขอมูลแลวเปลี่ยนเกรด จากนั้นจะรอรับรหัสนักศึกษาและวิชาใหม
// จนกวาจะกดปุม ESC เพื่อเลิกการทํ างาน

PROCEDURE R0301
   USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM= '1'
   DO WHILE LASTKEY() != 27
      GO TOP
      _IDSTD = 0
      _SUBJ = SPACE(8)
      @ 6,5 CLEAR
      @ 4,5 SAY "รหสันักศึกษาที่ขอเพิกถอนวิชา : " GET _IDSTD
      @ 5,5 SAY "รหสัวิชาที่ขอเพิกถอนวิชา : " GET _SUBJ
      READ
      LOCATE FOR IDSTD = _IDSTD .AND. SUBJ = _SUBJ
      IF FOUND()
         @ 6,5 SAY 'เกรดเดิม ' + STR(RGRADE)
         @ 7,5 SAY 'การขอเพิกถอนของ ' + STR(_IDSTD) + ' สมบูรณ'
         REPLACE RGRADE WITH 4.2
      ELSE
         @ 6,5 SAY 'ไมพบรหัสนักศึกษา ' + STR(_IDSTD)
      ENDIF
      INKEY(0)
   ENDDO
RETURN
: ตัวอยางที่ 5.26
&   ผลลัพธ
// พมิพตารางเรียนของนักศึกษาแตละวิชา
// รายงานแสดงใหเห็นวานักศึกษาแตละคน เรียนวิชาอะไร หอง เวลาและใครคือผูสอน
// ตวัอยางนี้แสดงการใช DEVPOS และ DEVOUT เพื่อน ําขอมูลมาแสดง
// เปนอีกวิธีหนึ่งที่จะน ําขอมูลออกทางเครื่องพิมพได
// แสดงใหเห็นการเชื่อมแฟมจาก 3 แฟมคือ TEACHER,SUBJECT และ REGIST
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE R0302
   DO SORTREG1   // จดัเรียงตามป ภาค และรหัส
   SELE 1 ; USE TEACHER
   SELE 2 ; USE SUBJECT

   SELE 3 ; USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
   GO TOP
   _IDSTDO = 0
   GASKPRT = ASKPRT()
   IF GASKPRT = 2
      SET DEVICE TO PRINTER
   ENDIF
   DO WHILE !EOF()
      IF IDSTD <> _IDSTDO
         IF GASKPRT = 2 ;  DEVPOS(PROW()+1,1)
            ELSE ; DEVPOS(ROW()+1,1) ; ENDIF
         DEVOUT(IDSTD)
         _IDSTDO = IDSTD
      ENDIF
      SELE 2 // แฟม SUBJECT.DBF
      LOCATE FOR SUBJ = REGIST->SUBJ
      IF FOUND()
         TXT = SUBJ + ROOM + STIME + STR(TEACHID)
         SELE 1  // แฟม TEACHER.DBF
         LOCATE FOR TEACHID = SUBJECT->TEACHID
         IF FOUND()
            TXT = TXT + TNAME + TSURN
         ENDIF
         IF GASKPRT = 2 ;  DEVPOS(PROW()+1,1)
            ELSE ; DEVPOS(ROW()+1,1) ; ENDIF
         DEVOUT(TXT) // น ําขอความทั้งหมดมารวมกันกอนพิมพ
      ENDIF
      SELE 3
      SKIP
   ENDDO
   IF GASKPRT = 2 ; EJECT ; ENDIF
   SET DEVICE TO SCREEN

   INKEY(0)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.27
&   ผลลัพธ
// พมิพรายงานผลการเรียนตลอดหลักสูตร โดยระบุรหัสนักศึกษา
// รายงานแสดงชื่อวิชา และเกรดที่นักศึกษาไดรับ
// บรรทัดสุดทั้งแสดงการนับเกรดที่นักศึกษาได A และ F
// แสดงใหเห็นวา สามารถใช SET FILTER แทนค ําสั่ง LOCATE ได
// และการพิมพขอมูลทางเครื่องพิมพดวย SET PRINT ON และ ?
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE R0303
   DO SORTREG1   // จดัเรียงตามป ภาค และรหัส
   SELE 1 ; USE IDSTD
   SELE 2 ; USE REGIST
      _IDSTD = 0
      @ 0,5 SAY "รหสันกัศึกษาที่ตองการรายงาน : " GET _IDSTD  ; READ
      GASKPRT = ASKPRT()
      IF  GASKPRT = 2 ; SET PRINT ON ;  ENDIF
      CLS ; _RYEAR = 0  ;  _RSEM = '0'
      SET FILTER TO IDSTD = _IDSTD
      COUNT TO CNTSUBJ  // น ําไปพิมพเปนผลลัพธตอนทายโปรแกรม
      COUNT TO CNTA   FOR RGRADE = 4  // นับเฉพาะเกรด A
      COUNT TO CNTF   FOR RGRADE = 0  // นับเฉพาะเกรด F
      GO TOP  // ค ําสัง่ COUNT ท ําใหตัวชี้เรคอรดเปลี่ยน จึงตองเซตตัวชี้ที่นี่
   SELE 1
      SET FILTER TO IDSTD = _IDSTD  ;    GO TOP
      ? IDSTD,INAME,ISURN,IMAJOR,ISTAT
   SELE 2
   DO WHILE !EOF() .AND. CNTSUBJ > 0
      IF RYEAR <> _RYEAR .OR. RSEM <> _RSEM
         ? RSEM , "/" , STR(RYEAR)
         _RYEAR = RYEAR
         _RSEM = RSEM

      ENDIF
      _GRADE = GRADECHR(RGRADE)
      ? " | " , SUBJ , " | " , _GRADE , " | "
      SKIP
   ENDDO
   ? "จ ํานวนวิชาที่นักศึกษาลงทะเบียน : " , STR(CNTSUBJ)
   ? "จ ํานวนวิชาที่ไดเกรด A : " , STR(CNTA)
   ? "จ ํานวนวิชาที่ไดเกรด F : " , STR(CNTF)
   IF GASKPRT = 2 ; EJECT ; ENDIF
   SET PRINT OFF
   INKEY(0)
RETURN
: ตัวอยางที่ 5.28
&   ผลลัพธ
// ปรบัปรงุแฟมอาจารย อยางงาย ๆ ดวยฟงกชัน BROWSE
// โดยเลือกกลุมขอมูลตามฟลดที่ตองการแกไขได
PROCEDURE R0304
   @ 1,5 SAY "เลอืกกลุมขอมูลที่ทานตองการปรับปรุง ตามฟลดใด ?"
   @ 2,5 PROMPT "1. รหัส"
   @ 3,5 PROMPT "2. ชื่อ"
   @ 4,5 PROMPT "3. สกุล"
   @ 5,5 PROMPT "4. อายุ"
   @ 6,5 PROMPT "5. เพศ"
   MENU TO FTEACH
   IF FTEACH > 0
      _KEYIN = "?                     "
      @ FTEACH + 1,20 SAY ":>" GET _KEYIN ; READ // รบัคาที่ตองการคนหา
      IF FTEACH = 4 .OR. FTEACH = 1
         _KEYCHK = VAL(_KEYIN)
      ELSE
         _KEYCHK = RTRIM(_KEYIN)
      ENDIF
   ENDIF
   USE TEACHER

   IF FTEACH = 1 ; SET FILTER TO TEACHID = _KEYCHK ; ENDIF
   IF FTEACH = 2 ; SET FILTER TO TNAME = _KEYCHK ; ENDIF
   IF FTEACH = 3 ; SET FILTER TO TSURN = _KEYCHK ; ENDIF
   IF FTEACH = 4 ; SET FILTER TO TAGE = _KEYCHK ; ENDIF
   IF FTEACH = 5 ; SET FILTER TO TSEX = _KEYCHK ; ENDIF
   COUNT TO CNTREC
   IF CNTREC > 0
      GO TOP
      BROWSE(7,5,21,75)
      PACK
   ELSE
      @ 7,5 SAY "ไมพบเรคอรด หรือกลุมเรคอรดที่ทานตองการปรับปรุง"
      INKEY(0)
   ENDIF
RETURN

5.4.5 โปรแกรมตาง ๆ ใน SUB04.PRG
โปรแกรมใน SUB04.PRG นี้มีโปรแกรมยอยทั้งหมด 5 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบงาน

ประจ ําภาคเรียน เชน โปรแกรมพิมพรายชื่อนักศึกษาติดหนาหองสอบในการสอบกลางภาค และปลายภาค โปรแกรม
พมิพรายชือ่นกัศกึษาแยกตามสถานภาพ หรือแยกตามสาขา  โปรแกรมพิมพใบรายงานผลการเรียนประจํ าภาคเรียน
โปรแกรมปรับปรุงเกรดนักศึกษาเมื่ออาจารยสงเกรด เปนตน

ระบบงานประจ ําภาคเรียน เปนระบบที่มีงานที่ตองทํ าในทุกภาคเรียน และสวนใหญเปนงานประจํ า เชน การ
จดัท ํารายงานแยกตามสถานภาพ การพิมพรายชื่อติดหนาหองสอบ หรือการปอนเกรดนักศึกษาเมื่อสิ้นภาคเรียน บาง
ครัง้ผูออกแบบระบบอาจน ําโปรแกรมในระบบอื่น ๆ มารวมในระบบนี้ได  เพราะระบบนี้เปนระบบใหญมีงานมาก แตผู
ออกแบบอาจแยกงานบางอยางไปสรางระบบยอยใหม  เพื่อท ําใหลดความซับซอนได

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB04.PRG ทั้งหมด 5 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.29 ถึง ตัวอยางที่ 5.33 ( R0401 - R0405 )
: ตัวอยางที่ 5.29
&   ผลลัพธ
// พมิพรายชื่อนักศึกษาแตละวิชาติดหนาหองสอบ
// รายงานแสดงล ําดับ รหัสวิชา รหัสนักศึกษา ชื่อ สกุล และเบอรเกาอี้ในหองสอบ
// โดยสมมติใหหองสอบกับหองเรียนใชหองเดียวกัน
PROCEDURE R0401
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG

   SELE 1 ; USE REGIST
      SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
      GO TOP
   SELE 2 ; USE IDSTD
   SELE 3 ; USE ROOM
   SELE 4 ; USE SUBJECT
   SELE 1
   _SUBJOLD = SPACE(8)
   IIDSTD := ISUBJ := 0
   DO WHILE !EOF()
      _IDSTD = IDSTD
      IIDSTD++
      IF SUBJ <> _SUBJOLD // ใชพิมพหัวของหนาแตละหนา
         ISUBJ++    // ใชนับวิชา
         IIDSTD = 1 // ใชนับนักศึกษาในวิชา
         INKEY(0.5)
         ? '*',ISUBJ,SUBJ
         _SUBJOLD = SUBJ
         SELE 4 // แฟมวิชา ใชคนหาหองเรียน
            LOCATE FOR SUBJ = REGIST->SUBJ
            IF FOUND()
               ?? ROOM  // หองสอบของวิชานั้น
            ENDIF
      ENDIF
      SELE 2 // แฟมนักศึกษา ใชคนหาชื่อ และสาขา
         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            ? IIDSTD,IDSTD,INAME,ISURN
         ELSE
            ? IIDSTD,_IDSTD  // เพราะรหัสนี้ไมมีใน IDSTD จึงตองใชตัวแปรแทน
         ENDIF
      SELE 3 // แฟมหองเรียน ใชคนหาเลขที่โตะ
         LOCATE FOR SUBJECT->ROOM = ROOM .AND. RCSEQ = IIDSTD

         IF FOUND()
            ?? "CHAIR : ",RCHAIR
         ENDIF
      SELE 1 // แฟมลงทะเบียน เปนแฟมหลักของรายงานนี้
      SKIP
   ENDDO
   INKEY(0)
   CLOSE ALL // ตอง CLOSE ALL เพราะเปดแฟมไวหลายพื้นที่
RETURN
: ตัวอยางที่ 5.30
&   ผลลัพธ
// พมิพรายชื่อแยกตามสถานภาพ โดยครั้งแรกแสดงผลทางจอภาพทันที
// ใชค ําสั่ง SET PRINT ON และ ? ในการจัดพิมพ
// ถากด P จะพิมพขอมูลที่พิมพไปแลวทางเครื่องพิมพทันที
// หากกด P อีกครั้งจะพิมพออกทางเครื่องพิมพอีก จนกวาจะกดปุมอื่น
PROCEDURE R0402
   DO SORTID2   // ใชจัดเรียงตามสถานภาพ และรหัส
   _ISTATO = " "
   USE IDSTD
   CHKPRT = "P"
   YESPRINTON = "N"
   DO WHILE CHKPRT = "P"
      WHILE !EOF()
         IF ISTAT <> _ISTATO
            ? "======================"
            ? "CURRENT STATUS : ",ISTAT
            _ISTATO = ISTAT
         ENDIF
         ? IDSTD,INAME,ISURN,IMAJOR
         SKIP
      END
      WAIT "ถาอยากพิมพใหกด P" // บรรทัดนี้ใหผลลัพธทางเครื่องพิมพ และจอภาพ
      CHKPRT = UPPER(CHR(LASTKEY()))
      IF CHKPRT = "P" // ตรวจสอบความตองการรายงานทางเครื่องพิมพ

         YESPRINTON = "Y"
         SET PRINT ON
         GO TOP ; CLS
      ENDIF
   ENDDO
   IF  YESPRINTON = "Y" ; EJECT ; ENDIF
   SET PRINT OFF
RETURN
: ตัวอยางที่ 5.31
&   ผลลัพธ
// พมิพรายชื่อแยกตามสาขา และจัดเรียงตามชื่อ
// จดักลุมขอมูลตามสาขา โดยรายงานแสดง รหัสนักศึกษา ชื่อ สกุล และสถานภาพ
// ผลลพัธแสดงทางเครื่องพิมพดวยการกด P เมื่อแสดงขอมูลทางจอภาพไปแลว
// จ ํานวนบรรทัดตอหนาของจอภาพระบุไวเปน 10 และเครื่องพิมพเปน 30
PROCEDURE R0403
   DO SORTID3  // ใชจัดเรียงตามสาขา และชื่อ
   _IMAJORO = 0
   USE IDSTD
   CHKPRT = "P"
   YESPRINTON = "N"
   LN = 1 ; LNSCR = 10 ; LNPRT = 30
   @ LN,1 SAY "ขอมลูชดุนี้เรียงตามจัดกลุมตามสาขา และเรียงตามชื่อ"
   DO WHILE CHKPRT = "P"
      WHILE !EOF()
         IF IMAJOR <> _IMAJORO
            LN++ ; @ LN,1 SAY "======================"
            LN++ ; @ LN,1 SAY "CURRENT MAJOR : " + STR(IMAJOR)
            _IMAJORO = IMAJOR
         ENDIF
         LN++ ; @ LN,1 SAY STR(IDSTD) + INAME + ISURN + ISTAT
         IF (YESPRINTON = "N" .AND. LN >= LNSCR) .OR. ;
            (YESPRINTON = "Y" .AND. LN >= LNPRT)
            INKEY(1)
            CLS // เงือ่นไขนี ้ท ําใหจํ านวนบรรทัดตอจอภาพ และเครื่องพิมพตางกัน

            LN = 0
         ENDIF
         SKIP
      END
      IF  YESPRINTON = "Y" ; EJECT ; ENDIF
      // ค ําสั่ง SET DEVICE TO PRINTER ไมสงผลตอค ําสั่ง WAIT
      WAIT "ถาอยากพิมพใหกด P" // สอบถามการพิมพออกเครื่องพิมพ
      CHKPRT = UPPER(CHR(LASTKEY()))
      IF CHKPRT = "P"
         YESPRINTON = "Y"
         SET DEVICE  TO PRINTER
         GO TOP ; CLS
      ENDIF
   ENDDO
   SET DEVICE TO SCREEN
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.32
&   ผลลัพธ
// ปรบัปรงุเกรดนักศึกษาเมื่ออาจารยสงเกรด โดยระบุปการศึกษา ภาค และวิชา
// รอรบัเกรดใหม แลวปรับปรุงทีละคน จนหมดขอมูลที่ตรงกับเงื่อนไข
PROCEDURE R0404
   DO SORTREG1 // ใชจัดเรียงขอมูลตามป ภาค รหัส และวิชา
   _RYEAR = 1996
   _RSEM = '1'
   _SUBJ = SPACE(8)
   USE REGIST
   DO WHILE !FOUND()
      @ 5,5 SAY "ปรบัปรุงเกรดนักศึกษา"
      @ 6,5 SAY "ปการศึกษา " GET _RYEAR PICT "9999"
      @ 7,5 SAY "ภาคการศึกษา " GET _RSEM
      @ 8,5 SAY "วชิาที่ตองการปอนเกรด " GET _SUBJ PICT "@!"
      READ
      LOCATE FOR _RYEAR = RYEAR .AND. _RSEM = RSEM .AND. SUBJ = _SUBJ

   ENDDO
   SET FILTER TO RYEAR = _RYEAR .AND. RSEM = _RSEM .AND. SUBJ = _SUBJ
   COUNT TO CNTALL
   GO TOP
   CNT = 1
   @ 9,5 SAY "จ ํานวนนกัศึกษาทั้งหมดในวิชานี้ : "+STR(CNTALL)
   DO WHILE !EOF() .AND. LASTKEY() != 27
      _GRADE = "  "
      WHILE _GRADE = "  "
         @ 10,5 SAY STR(CNT) +": "+ STR(IDSTD) +" : "+ STR(RGRADE) +": ";
            GET _GRADE
         READ
      END
      _RGRADE = GRADENUM(_GRADE) // แปลงเกรดตัวอักษรเปนตัวเลข
      REPLACE RGRADE  WITH _RGRADE
      SKIP
      CNT++
   ENDDO
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.33
&   ผลลัพธ
// พมิพใบรายงานผลการเรียนประจ ําภาค โดยระบุปการศึกษา และภาคการศึกษา
// จดักลุมตามรหัสนักศึกษา เพื่อพิมพใบรายงานผลการเรียนใหนักศึกษาแตละคน
// รายงานแสดงล ําดับ รหัสวิชา และเกรด
// บรรทดัสดุทายแสดงเกรดเฉลี่ยประจ ําภาคเรียนของนักศึกษาคนนั้น
PROCEDURE R0405
   _RYEAR = 1996
   _RSEM = '1'
   @ 5,5 SAY "ปการศึกษา" GET _RYEAR PICT "9999"
   @ 6,5 SAY "ภาคการศึกษา" GET _RSEM
   READ
   USE REGIST
   SET FILTER TO RYEAR = _RYEAR .AND. RSEM = _RSEM

   GO TOP
   _GPA = 0
   _CNTSUBJ = 0
   _SUMPOINT = 0
   _IDSTDO = 0
   CNTID = 0
   CNT = 1
   DO WHILE !EOF() .AND. LASTKEY() != 27
      IF IDSTD <> _IDSTDO
         IF CNT <> 1
            _NEXTIDSTD = IDSTD
            COUNT TO _CNTSUBJ FOR RYEAR = _RYEAR .AND. RSEM = _RSEM .AND.;
              IDSTD = _IDSTDO .AND. RGRADE <= 4
            SUM RGRADE TO _SUMPOINT FOR RYEAR = _RYEAR .AND. RSEM = _RSEM ;
              .AND. IDSTD = _IDSTDO .AND. RGRADE <= 4
            _GPA  = _SUMPOINT / _CNTSUBJ
            @ ROW()+1,5 SAY "เกรดเฉลี่ยประจ ําภาคเรียนนี้คือ "+STR(_GPA)
            INKEY(3)
            LOCATE FOR IDSTD = _NEXTIDSTD // เพราะ COUNT จะเปลี่ยนตัวชี้ใหม
         ENDIF  // จงึตองใช LOCATE เลื่อนตัวชี้ใหถูกตองอีกครั้ง
         CNT = 1
         CNTID++
         @ 7,5 CLEAR TO 22,75
         SETPOS(7,1)
         @ ROW()+1,5 SAY STR(CNTID) + " : " + STR(IDSTD)
         @ ROW()+1,5 SAY "================================"
         _IDSTDO = IDSTD
      ENDIF
      _GRADE = GRADECHR(RGRADE)  //  แปลงเกรดตัวเลขเปนตัวอักษร
      @ ROW()+1,10 SAY STR(CNT) + " : " + SUBJ + " : " + _GRADE
      SKIP
      CNT++
   ENDDO

   CLOSE ALL
RETURN

5.4.6 โปรแกรมตาง ๆ ใน SUB05.PRG
โปรแกรมใน SUB05.PRG นี้มีโปรแกรมยอยทั้งหมด 5 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบปรับปรุง

และพมิพขอมูลจากแฟม เชน โปรแกรมปรับปรุงแฟมและพิมพ ของแฟมในแบบตาง ๆ  ท ําใหสามารถเลือกเปดแฟมใน
ระบบ โดยระบบประกอบดวย แฟมนักศึกษา แฟมวิชา แฟมผูสอน แฟมลงทะเบียน และแฟมหอง

ระบบงานปรบัปรุงและพิมพจากแฟมอยางงาย ๆ ชวยทํ าใหการปรับปรุงขอมูลทํ าไดทันท ีแตไมเหมาะกับการ
น ําไปใชงานจรงิ เพราะการปรับปรุงขอมูลจะทํ าใหควบคุมความปลอดภัยหรือความถูกตองทํ าไดยาก ดังนั้นการปรับ
ปรงุจงึควรผานโปรแกรมที่อนุญาติใหแกไขขอมูล โดยมีการตรวจสอบความถูกตองของขอมูลอยางดี กอนจะจัดเก็บขอ
มลูลงแฟม  และการพิมพขอมูลควรที่จะเชื่อมขอมูลกับแฟมตาง ๆ ท ําใหเกิดรายงานทีอ่านเขาใจไดโดยงายเพราะการ
น ําแฟมเพยีงแฟมเดียวมาพิมพจะไมสมบูรณนัก เพราะมีแฟมหลาย ๆ แฟมที่เก็บขอมูลเปนรหัส  ท ําใหตองนํ ารหัสนั้น
ไปเชือ่มกับแฟมอื่น จึงจะไดขอมูลที่อานแลวเขาใจไดทันที

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB05.PRG ทั้งหมด 5 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.34 ถึง ตัวอยางที่ 5.38 ( R0501 - R0505 )
: ตัวอยางที่ 5.34
&   ผลลัพธ
// การปรับปรุงแฟมนักศึกษา
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0501
   _IDSTD = 0
   X = ALERT("ปรบัปรงุแฟมนักศึกษา",{'พิมพขอมูล','เพิ่ม','ลบ','แกไข'})
   DO CASE
      CASE X = 0 ; WAIT "ไมไดเลือกตัวเลือกใดเลย"
      CASE X = 1
      // พมิพขอมูลจากแฟมนักศึกษา
         @ 10,5 SAY "พมิพขอมูลจากแฟมนักศึกษา"
         GASKPRT = ASKPRT()
         IF GASKPRT = 0 ; RETURN ; ENDIF
         USE IDSTD
         IF GASKPRT = 1
            CLS
            LIST IDSTD,INAME,ISURN,IMAJOR,ISTAT
            INKEY(0)

         ENDIF
         IF GASKPRT = 2        // ผลออกทั้งจอภาพและเครื่องพิมพ
            LIST IDSTD,INAME,ISURN,IMAJOR,ISTAT TO PRINTER
            EJECT
         ENDIF
      CASE X = 2
      // เพิ่มขอมูลในแฟมนักศึกษา
         USE IDSTD
         APPEND BLANK
         @ 10,5 SAY "เพิม่ขอมูลในแฟมนักศึกษาใหม"
         @ 11,5 SAY "รหัส" GET IDSTD
         @ 12,5 SAY "ชือ่" GET INAME
         @ 13,5 SAY "สกุล" GET ISURN
         @ 14,5 SAY "สาขา" GET IMAJOR
         @ 15,5 SAY "สถานภาพ" GET ISTAT
         READ
         REPLACE IDSTD WITH IDSTD ,INAME WITH INAME ,ISURN WITH ISURN,;
                 IMAJOR WITH IMAJOR,ISTAT WITH ISTAT
      CASE X = 3
      // ลบขอมูลจากแฟมนักศึกษา
         @ 10,5 SAY "ลบขอมูลจากแฟมนักศึกษา"
         USE IDSTD
         @ 11,5 SAY "รหสันักศึกษาที่ตองการลบ : " GET _IDSTD ; READ
         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            DELETE
            PACK
         ENDIF
      CASE X = 4
      // แกไขขอมูลจากแฟมนักศึกษา
         @ 10,5 SAY "แกไขขอมูลจากแฟมนักศึกษา"
         USE IDSTD
         @ 11,5 SAY "รหสันักศึกษาที่ตองการแกไข : " GET _IDSTD ; READ

         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            @ 12,5 SAY "ชื่อ" GET INAME
            @ 13,5 SAY "สกุล" GET ISURN
            @ 14,5 SAY "สาขา" GET IMAJOR
            @ 15,5 SAY "สถานภาพ" GET ISTAT
            READ
            REPLACE IDSTD WITH _IDSTD ,INAME WITH INAME ,ISURN WITH ISURN,;
                 IMAJOR WITH IMAJOR,ISTAT WITH ISTAT
         ENDIF
   ENDCASE
RETURN
: ตัวอยางที่ 5.35
&   ผลลัพธ
// การปรับปรุงแฟมวิชา
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0502
   _SUBJ = SPACE(8)
   _TEACHID = 0
   _ROOM = SPACE(4)
   _STIME = SPACE(11)
   _SREGIST := _SLIMIT := 0
   X = ALERT("ปรบัปรุงแฟมวิชา",{'พิมพขอมูล','เพิ่ม','ลบ','แกไข'})
   DO CASE
      CASE X = 0 ; WAIT "ไมไดเลือกตัวเลือกใดเลย"
      CASE X = 1
      // พมิพขอมูลจากแฟมวิชา
         @ 4,5 SAY "พมิพขอมูลจากแฟมวิชา"
         GASKPRT = ASKPRT()
         IF GASKPRT = 0 ; RETURN ; ENDIF
         USE SUBJECT
         IF GASKPRT = 1
            AR = {"SUBJ","TEACHID","ROOM","STIME","SREGIST","SLIMIT"}
            DBEDIT(5,5,20,75,AR)

         ENDIF
         IF GASKPRT = 2        // ผลออกทั้งจอภาพและเครื่องพิมพ
            LIST SUBJ,TEACHID,ROOM,STIME,SREGIST,SLIMIT
            EJECT
         ENDIF
      CASE X = 2
      // เพิ่มขอมูลในแฟมวิชา
         USE SUBJECT
         @ 10,5 SAY "เพิม่ขอมูลในแฟมวิชาใหม"
         @ 11,5 SAY "รหัสวิชา" GET _SUBJ
         @ 12,5 SAY "รหัสอาจารยผูสอน" GET _TEACHID
         @ 13,5 SAY "หองเรียน" GET _ROOM
         @ 14,5 SAY "เวลาเรียน" GET _STIME
         @ 15,5 SAY "จ ํานวนที่ลงทะเบียน" GET _SREGIST
         @ 16,5 SAY "จ ํานวนสูงสุดที่อนุญาต"ิ GET _SLIMIT
         READ
         IF ASKUPD() = 1
            APPEND BLANK
            REPLACE SUBJ WITH _SUBJ,TEACHID WITH _TEACHID,ROOM WITH _ROOM,;
                    STIME WITH _STIME,SREGIST WITH _SREGIST,SLIMIT WITH _SLIMIT
         ENDIF
      CASE X = 3
      // ลบขอมูลจากแฟมวิชา
         @ 10,5 SAY "ลบขอมูลจากแฟมวิชา"
         USE SUBJECT
         @ 11,5 SAY "รหสัวิชาที่ตองการลบ : " GET _SUBJ ; READ
         LOCATE FOR SUBJ = _SUBJ
         IF FOUND()
            @ 12,5 SAY SUBJ
            @ 13,5 SAY TEACHID
            @ 14,5 SAY ROOM
            @ 15,5 SAY STIME
            @ 16,5 SAY SREGIST

            @ 17,5 SAY SLIMIT
            IF ASKUPD() = 1
               DELETE ; PACK
            ENDIF
         ENDIF
      CASE X = 4
      // แกไขขอมูลจากแฟมวิชา
         @ 10,5 SAY "แกไขขอมูลจากแฟมวิชา"
         USE SUBJECT
         @ 11,5 SAY "รหัสวิชา : " GET _SUBJ ; READ
         LOCATE FOR SUBJ = _SUBJ
         IF FOUND()
            _SUBJ = SUBJ
            _TEACHID = TEACHID
            _ROOM = ROOM
            _STIME = STIME
            _SREGIST = SREGIST
            _SLIMIT = SLIMIT
            @ 11,5 SAY "รหัสวิชา : " GET _SUBJ
            @ 12,5 SAY "รหัสอาจารยผูสอน" GET _TEACHID
            @ 13,5 SAY "หองเรียน" GET _ROOM
            @ 14,5 SAY "เวลาเรียน" GET _STIME
            @ 15,5 SAY "จ ํานวนที่ลงทะเบียน" GET _SREGIST
            @ 16,5 SAY "จ ํานวนสูงสุดที่อนุญาต"ิ GET _SLIMIT
            READ
            IF ASKUPD() = 1
               REPLACE SUBJ WITH _SUBJ,TEACHID WITH _TEACHID,;
                       ROOM WITH _ROOM,STIME WITH _STIME,;
                       SREGIST WITH _SREGIST,SLIMIT WITH _SLIMIT
            ENDIF
         ENDIF
   ENDCASE
RETURN

: ตัวอยางที่ 5.36
&   ผลลัพธ
// การปรับปรุงแฟมผูสอน
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0503
   USE TEACHER
   BROWSE(5,5,22,75)
   PACK
RETURN
: ตัวอยางที่ 5.37
&   ผลลัพธ
// การปรับปรุงแฟมลงทะเบียน
// โดยเลือกจัดเรียงขอมูลดวยค ําสั่ง SORT
// ปรบัปรุงขอมูลโดยใชคํ าสั่ง BROWSE
PROCEDURE R0504
   @ 4,5 SAY "การปรับปรุงแฟมลงทะเบียน"
   @ 9,5 SAY "สัง่จัดเรียงกอนปรับปรุงขอมูล"
   @ 10,5 PROMPT "1. จดัเรยีงตามป ภาค รหัสนักศึกษา และวิชา"
   @ 11,5 PROMPT "2. จดัเรยีงตามวิชา ป ภาค และรหัสนักศึกษา"
   @ 12,5 PROMPT "3. จดัเรยีงตามรหัสนักศึกษา ป ภาค และวิชา"
   @ 13,5 PROMPT "4. จดัเรยีงตามเกรด ป ภาค วิชา และรหัสนักศึกษา"
   @ 14,5 PROMPT "5. จดัเรยีงตามรหัสนักศึกษา และวิชา จากมากไปนอย"
   @ 15,5 PROMPT "6. ไมจัดเรียงใหม"
   MENU TO _SORTORDER
   IF _SORTORDER > 0 .AND. _SORTORDER < 6
      USE REGIST
      DO CASE
         CASE _SORTORDER = 1
            SORT ON RYEAR,RSEM,IDSTD,SUBJ TO TEMPF
         CASE _SORTORDER = 2
            SORT ON SUBJ,RYEAR,RSEM,IDSTD TO TEMPF
         CASE _SORTORDER = 3
            SORT ON IDSTD,RYEAR,RSEM,SUBJ TO TEMPF
         CASE _SORTORDER = 4
            SORT ON GRADE,RYEAR,RSEM,SUBJ,IDSTD TO TEMPF

         CASE _SORTORDER = 5
            SORT ON IDSTD/D ,SUBJ TO TEMPF
       ENDCASE
       USE TEMPF
       COPY TO REGIST
       CLOSE
   ENDIF
   USE REGIST
   BROWSE(5,5,22,75)
   PACK
RETURN
: ตัวอยางที่ 5.38
&   ผลลัพธ
// การปรับปรุงแฟมหอง
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0505
   _ROOM = SPACE(4)
   _RCSEQ = 0
   _RCHAIR = SPACE(2)
   X = ALERT("ปรบัปรุงแฟมหอง",{'พิมพขอมูล','เพิ่ม','ลบ','แกไข'})
   DO CASE
      CASE X = 0 ; WAIT "ไมไดเลือกตัวเลือกใดเลย"
      CASE X = 1
      // พมิพขอมูลจากแฟมหอง
         @ 4,5 SAY "พมิพขอมูลจากแฟมหอง"
         GASKPRT = ASKPRT()
         IF GASKPRT = 0 ; RETURN ; ENDIF
         USE ROOM
         IF GASKPRT = 1
            DISPLAY ROOM,RCSEQ,RCHAIR ALL
            INKEY(5)
         ENDIF
         IF GASKPRT = 2        // ผลออกทั้งจอภาพและเครื่องพิมพ
            FOR I = 1 TO  RECCOUNT()

               DISPLAY I,ROOM,RCSEQ,RCHAIR TO PRINTER
               SKIP
            NEXT
            EJECT
         ENDIF
      CASE X = 2
      // เพิ่มขอมูลในแฟมหอง
         USE ROOM
         @ 10,5 SAY "เพิม่ขอมูลในแฟมหองใหม"
         SET FORMAT TO R0505FORM
         READ
         SET FORMAT TO
         IF ASKUPD() = 1
            APPEND BLANK
            REPLACE ROOM WITH _ROOM,RCSEQ WITH _RCSEQ,RCHAIR WITH _RCHAIR
         ENDIF
      CASE X = 3
      // ลบขอมูลจากแฟมหอง
         @ 9,5 SAY "ลบขอมูลจากแฟมหอง"
         USE ROOM
         @ 10,5 SAY "รหสัหองที่ตองการลบ : " GET _ROOM
         @ 11,5 SAY "ล ําดับเกาอี้ : " GET _RCSEQ
         READ
         LOCATE FOR ROOM = _ROOM .AND. RCSEQ = _RCSEQ
         IF FOUND()
            @ 12,5 SAY RCHAIR
            IF ASKUPD() = 1
               DELETE ; PACK
            ENDIF
         ENDIF
      CASE X = 4
      // แกไขขอมูลจากแฟมหอง
         @ 7,5 SAY "แกไขขอมูลจากแฟมหอง"

         USE ROOM
         @ 8,5 SAY "รหสัหองที่ตองการแกไข : " GET _ROOM
         @ 9,5 SAY "ล ําดับเกาอี้ : " GET _RCSEQ
         READ
         LOCATE FOR ROOM = _ROOM .AND. RCSEQ = _RCSEQ
         IF FOUND()
            _ROOM = ROOM
            _RCSEQ = RCSEQ
            _RCHAIR = RCHAIR
            SET FORMAT TO R0505FORM
            READ
            SET FORMAT TO    // ตองยกเลิก FORMAT เพราะปองกันการ READ แบบอื่น
            IF ASKUPD() = 1
               REPLACE ROOM WITH _ROOM,RCSEQ WITH _RCSEQ,RCHAIR WITH _RCHAIR
            ENDIF
         ENDIF
   ENDCASE
   PROCEDURE R0505FORM
   @ 10,5 SAY "หองเรียน" GET _ROOM
   @ 11,5 SAY "ล ําดับของเกาอี้" GET _RCSEQ
   @ 12,5 SAY "เลขที่เกาอี"้ GET _RCHAIR
   RETURN
RETURN

